

33 New APPLE

**Computer Programs
for Home, School & Office**

A handy collection of ready-to-run software for
businessmen, teachers, students and hobbyists.



by Fred White

33 New APPLE

**Computer Programs
For Home, School & Office**

33 New APPLE

**Computer Programs
For Home, School & Office**

by Fred White

ARCsoft Publishers

WOODSBORO, MARYLAND 21798

**FIRST EDITION
SECOND PRINTING**

© 1982 by ARCsoft Publishers, P.O. Box 132, Woodsboro, MD 21798 USA

Printed in the United States of America

Reproduction or publication of the contents of this book, in any manner, without express permission of the publisher, is prohibited. No liability is assumed with respect to the use of the information herein.

Trademark credits and software copyrights:

APPLE is a trademark of Apple Computer, Inc.

Programs in this book are copyright 1982 by ARCsoft Publishers.

ISBN 0-86668-016-0

Preface

APPLE may be the world's most popular computer system. Its lightweight desktop design and powerful language capability have placed it at the forefront of the new wave of personal computers for home, school and office.

Not at all a toy, its hardware and software combination have made it a highly useful tool in the business environment and the classroom as well as for practical jobs around the home.

The number of jobs which the APPLE computer can handle is limited only by the scope of the imagination. In this book we have attempted to create 33 new specific practical sets of applications software for your use. In fact, you'll find 35 different programs in this volume.

This book is written for newcomers and beginners, as well as for advanced users of microcomputers. Our intention has been to provide ready-to-run programs. You type them in and the APPLE does the rest.

This book is a companion volume to *101 APPLE Computer Programming Tips & Tricks*.

— Fred White

Table of Contents

8 Introduction

Programs for Businessmen

- 12 Profit Estimator**
- 16 Inventory Review**
- 20 Invoice Computer**
- 21 Pricing Mark-Up**
- 25 Daily Codes**
- 26 Hourly Wages**
- 28 Highest/Lowest Item Price**
- 29 Salesman's Commission**
- 31 Employee Roll Sorter**
- 32 Unit Price**
- 33 Executive Decision Maker**

Programs for Students and Teachers

- 36 Math Flasher**
- 38 Exam Score Sorting**
- 39 U.S. Presidents**
- 41 Number of Days In a Month**
- 42 Foreign Capitals**
- 45 Weights & Measures**

- 47 State Geographic Centers
- 49 Analyzing Test Evaluation
- 53 Lakes of the World
- 55 Deserts of the World
- 57 Volcanoes of the World
- 58 Chemistry: Elements
- 62 Chemistry: Atomic Number

Programs for the Home

- 68 Interest On Money
- 72 Club Member List
- 73 Electric Bill Analysis
- 79 High/Low Bowling Score
- 80 Funny Similes
- 81 Recall Tester
- 83 Event Timer
- 84 Photography: Flash Exposure
- 85 Photography: Close-Ups
- 86 Savings Quickie
- 87 Random Number Quality Checker

Appendices

- 89 BASIC Words in the APPLE
- 92 APPLE Error Messages

Introduction

There's a great need for practical, useful software for the new generation of popular, personal computers. The APPLE is powerful and flexible and versatile — but what can it do? The aim of this book is to provide at least 33 (in fact, there are 35 different programs in this book) complete ready-to-run sets of program lists for you to use in making your APPLE computer work for you.

These are designed to be typed in, just as you find them in this book, with no other programming needed. You really don't have to be a computer programmer to use these pieces of software.

This book can be used by newcomers and beginners as well as old-timers in the programming game. Each of the 35 programs in this book has been thoroughly tested and run many times on an APPLE II computer. Each is ready to use right away.

The book has been organized into three sections plus an appendix: programs for the business office; programs for the classroom; and practical programs for the home. The appendices give you a handy place to look for the BASIC words and error messages used in your APPLE.

Naturally, the sections of this book are not rigid dividing lines. You probably will find something in the business section or classroom section applicable to your home use. Or you may take one of the home programs to work or school. Try them all! They're great fun to run. They are especially designed to be short so you won't have to spend hours typing them into your APPLE.

Other computers

These programs will run on other microcomputer systems, but you will have to make minor modifications to program lines in some cases. Graphics commands, especially, will differ elsewhere.

Also, if you use other than APPLE II, line numbering, logical tests, multiplication symbols, print statements and other instructions may differ. If you want to run these programs on a machine, other than an APPLE II, better check the list of available BASIC words and how they work on your computer before running.

REMARKS

As you read through the 35 programs in this book, you will notice few REMARKS. The author's training in writing BASIC-language computer programs included an emphasis on brevity and saving memory space. A sharp editing pencil was in order — and still is!

Remarks and explanations in the software were out. Honing, fine-tuning and waste trimming were in. Use of coding form program-writing worksheets, such as the *APPLE BASIC Coding Form* (published by *ARCsoft Publishers*), was encouraged. The objective always was — and still is — to make the most efficient use of available memory.

Always remember: even though they may be headed toward the same goal, no two programmers will write exactly the same list of BASIC instructions or program lines from scratch. As you load these 35 programs into your APPLE, you'll make modifications to suit your personal needs and interests. Exact wording of PRINT statements, for instance, can be changed. Or two or more programs could be combined into one grand scheme. Your applications may vary.

ENTER vs. ENTER

We use in our write-ups the word RETURN and the word ENTER interchangeably. In either case we mean the large key marked RETURN on the right-hand side of the APPLE keyboard.

By the way, the author would like to hear of improvement suggestions as well as ideas for future volumes in this series for the APPLE computer. He may be addressed in care of *ARCsoft Publishers*, Post Office Box 132, Woodsboro, Maryland 21798 USA.

If you want to load more than one of these programs into your computer at the same time, be sure to use different sets of line numbers for different programs.

Programs for Businessmen

Profit Estimator

How much cash flow will I generate if I sell 100 thingamabobs? A question faced everyday in the business office. Whether you sell large lots at wholesale, small quantities across the retail counter, or individual items via mail order, this program will give you a quick estimate of expected cash flow and potential profits. It allows fast comparisons when quick decisions are needed.

The computer asks you questions about the quantity of items involved, prices, quantity sold, discounts, etc. Then it calculates unit price, unit profit, gross profit, return percentages, sales needed to break even, and more, depending upon which part of the program you are using.

The program is divided into four main blocks:

- an opening billboard from line 10 through line 210;
- wholesale computations, lines 1000 to 1320;
- mail order computations, lines 2000 to 2360;
- ad response computations, lines 3000 to 3590.

Remember that the program only *estimates*, it is not exact. The wholesale, direct-mail, or ad-response manufacturing cost asked by the computer is *total*, not per unit.

This program is a useful tool for small business, whether a local furniture store, supermarket, wholesaler, or regional mail-order house.

Program Listing

```
10 HOME: CLEAR
20 PRINT "PROFIT ESTIMATOR"
30 PRINT
40 PRINT "WHICH TYPE OF SALE:"
50 PRINT "WHOLESALE (W)"
60 PRINT "DIRECT MAIL (D)"
70 PRINT "MEDIA AD RESPONSE (M)"
80 PRINT
100 PRINT CHR$(7)
110 PRINT "W,D, OR M ?"
120 GET Q$
130 IF Q$="W" THEN 1000
```

```

140 IF Q$="D" THEN 2000
150 IF Q$="M" THEN 3000
170 HOME
180 PRINT "OKAY. YOU SELECTED ";Q$
190 PRINT "HOWEVER, ";Q$;" IS NOT"
200 PRINT "A CHOICE. TRY AGAIN"
210 GOTO 40
1000 HOME: CLEAR
1010 PRINT "WHOLESALE"
1020 PRINT "PROFIT ESTIMATOR"
1030 PRINT
1050 PRINT CHR$(7)
1060 PRINT "PLEASE ANSWER THESE QUESTIONS"
1070 PRINT
1080 INPUT "MANUFACTURING COST $";C
1090 PRINT CHR$(7)
1100 INPUT "QUANTITY MANUFACTURED";P
1110 PRINT CHR$(7)
1120 INPUT "LIST PRICE OF ITEM";L
1130 PRINT CHR$(7)
1140 INPUT "TOTAL QUANTITY SOLD";S
1150 IF S<>0 THEN 1160
1155 GOTO 10
1160 INPUT "WHOLESALE DISCOUNT %";D
1170 UC=C/P
1180 UP=(L*((100-D)/100))-UC
1190 W=L*S*((100-D)/100)
1200 G=W-(S*UC)
1210 PRINT CHR$(7)
1220 PRINT
1230 PRINT "-----"
1240 PRINT "UNIT COST IS $";UC
1250 PRINT "UNIT PROFIT IS $";UP
1260 PRINT "WHOLESALE GROSS $";W
1270 PRINT "WHOLESALE PROFIT $";G
1280 PRINT "-----"
1290 PRINT
1300 PRINT "FOR MORE,PRESS ANY KEY"
1310 GET A$
1320 GOTO 10
2000 HOME: CLEAR
2010 PRINT "DIRECT MAIL"

```

```

2020 PRINT "PROFIT ESTIMATOR"
2030 PRINT
2050 PRINT CHR$(7)
2060 PRINT"PLEASE ANSWER THESE QUESTIONS"
2070 PRINT
2080 INPUT"MANUFACTURING COST $";C
2090 PRINT CHR$(7)
2100 INPUT"QUANTITY MANUFACTURED";P
2110 PRINT CHR$(7)
2120 INPUT"LIST PRICE OF ITEM";L
2130 PRINT CHR$(7)
2140 INPUT "TOTAL QUANTITY SOLD";S
2150 PRINT CHR$(7)
2160 INPUT"FLYERS MAILED";K
2170 PRINT CHR$(7)
2180 INPUT"FLYER PRINT COST$";R
2190 PRINT CHR$(7)
2200 INPUT"POSTAGE COST $";M
2210 UC=C/P
2220 J=100*S/K
2230 T=L*S-(R+M+UC*S)
2240 U=L*S
2250 PRINT CHR$(7)
2260 PRINT
2270 PRINT "-----"
2280 PRINT"DIRECT MAIL"
2290 PRINT"RETURN IS ";J;" PERCENT"
2300 PRINT"DIRECT MAIL GROSS $";U
2310 PRINT"DIRECT MAIL PROFIT $";T
2320 PRINT "-----"
2330 PRINT
2340 PRINT"FOR MORE,PRESS ANY KEY"
2350 GET A$
2360 GOTO 10
3000 HOME: CLEAR
3010 PRINT "AD RESPONSE"
3020 PRINT "PROFIT ESTIMATOR"
3030 PRINT
3050 PRINT CHR$(7)
3060 PRINT"PLEASE ANSWER THESE QUESTIONS"
3070 PRINT
3080 INPUT"MANUFACTURING COST $";C

```



```

3090 PRINT CHR$(7)
3100 INPUT"QUANTITY MANUFACTURED";P
3110 PRINT CHR$(7)
3120 INPUT"LIST PRICE OF ITEM";L
3130 PRINT CHR$(7)
3140 INPUT"AD COST PER INSERTION";A
3150 PRINT CHR$(7)
3160 INPUT"NUMBER OF INSERTIONS";I
3170 PRINT CHR$(7)
3180 HOME
3190 PRINT"WHICH DO YOU WANT TO KNOW?"
3200 PRINT
3210 PRINT"SALES QUANTITY NEEDED"
3220 PRINT"TO BREAK EVEN (Q)"
3230 PRINT
3240 PRINT"PROFIT FROM SELLING"
3250 PRINT"A SPECIFIC QUANTITY (P)"
3260 PRINT
3270 PRINT"P OR Q ?"
3280 GET A$
3300 IF A$="P" THEN 3460
3310 IF A$="Q" THEN 3330
3320 GOTO 3280
3330 PRINT
3340 B=INT((C+A*I)/L)+1
3350 PRINT "-----"
3360 PRINT"SELL ";B;" TO BREAK EVEN"
3370 PRINT"INCLUDING COVERING"
3380 PRINT"$";C;" MANUFACTURING COST"
3390 PRINT"AND $";A*I;" AD CAMPAIGN"
3400 PRINT "-----"
3410 PRINT
3420 PRINT" FOR MORE,PRESS ANY KEY"
3430 GET A$
3450 GOTO 10
3460 PRINT
3470 INPUT"QUANTITY SOLD";S
3480 N=S*L
3490 UC=C/P
3500 E=S*L-S*UC-A*I
3510 PRINT
3520 PRINT "-----"

```

```
3530 PRINT"ORDERS GROSS IS $";N
3540 PRINT"DIRECT MAIL PROFIT IS $";E
3550 PRINT "-----"
3560 PRINT
3570 PRINT"FOR MORE,PRESS ANY KEY"
3580 GET A$
3590 GOTO 10
```

Inventory Review

Here's a time-saver, useful for small businesses in evaluating the contents of inventory. As written, it can handle 50 different inventory items. To change that quantity, increase or decrease the number 50 in lines 10 and 100 and the number 49 in line 430.

The computer will ask you the name of a product's manufacturer, what the product is, its retail price, and the quantity on hand in inventory.

It sorts all such information on the various items in inventory and lists, alphabetically by manufacturer, and then lists again, alphabetically by product name.

Note in the sample run, the program can tell the difference between a manufacturer's similar products and between the same products from different manufacturers. If one manufacturer has the same product available at different prices, the computer sorts by price. See the sample run for an example of this.

Use a printer and you will be able to print-out for your permanent records an alphabetized list of all products in inventory and a summary review of quantities, prices, and values.

Whether you display results on your television screen or on a printer, statistics include total number of different items in inventory; number of different items in various price ranges; and total retail value of the inventory.

You can change the price ranges by modifying the less-than values in lines 170-210.

Program Listing

```
10 HOME: DIM M$(50)
20 PRINT "*****"
30 PRINT "* INVENTORY REVIEW *"
40 PRINT "*****"
50 PRINT: PRINT: PRINT CHR$(7)
100 FOR L=1 TO 50
110 INPUT "MANUFACTURER: "; P$
120 IF P$="" THEN 400
130 INPUT "PRODUCT: "; Q$
140 INPUT "RETAIL PRICE: $"; G$
150 INPUT "QUANTITY ON HAND: "; GG$
160 X=VAL(G$): XX=VAL(GG$)
170 IF X<10 THEN F=F+1: GOTO 220
180 IF X<20 THEN D=D+1: GOTO 220
190 IF X<30 THEN C=C+1: GOTO 220
200 IF X<40 THEN B=B+1: GOTO 220
210 A=A+1
220 Y=Y+X
230 Z=Z+1
240 IF Z=1 THEN J=X: K=X
250 IF X<J THEN J=X
260 IF X>K THEN K=X
270 I=I+X
280 II=XX*X: TT=TT+II
290 H=I/Z
300 M$(L)=P$+" "+Q$+" "+ "$"+G$+" "+GG$
310 PRINT
320 NEXT L
400 HOME: PRINT CHR$(7)
410 PRINT "SORTING"
420 T=0
430 FOR L=1 TO 49
440 IF M$(L)<=M$(L+1) THEN 460
450 E$=M$(L): M$(L)=M$(L+1): M$(L+1)=E$: T=T+1
460 NEXT L
470 IF T=1 THEN 420
500 HOME: PRINT CHR$(7)
510 PRINT "MFG / ITEM / PRICE / QUANTITY"
520 PRINT
530 FOR L=1 TO 50
```

```

540 IF M$(L)<>" " THEN PRINT M$(L)
550 NEXT L
560 PRINT:PRINT
570 PRINT"TO REVIEW INVENTORY, PRESS ANY KEY"
580 GET KY$
700 HOME:PRINT CHR$(7)
710 PRINT TAB(10)"INVENTORY REVIEW"
720 PRINT:PRINT"NUMBER DIFFERENT ITEMS:"
730 PRINT A;" FROM $40.00 ON UP"
740 PRINT B;" FROM $30.00 TO 39.99"
750 PRINT C;" FROM $20.00 TO 29.99"
760 PRINT D;" FROM $10.00 TO 19.99"
770 PRINT F;" UNDER $10.00"
780 PRINT:PRINT"TOTAL ";Z;" ITEMS"
790 PRINT:PRINT"AVERAGE PRICE $";H
800 PRINT"RANGE $";J;" TO $";K
810 PRINT"TOTAL RETAIL VALUE $";TT
820 PRINT:PRINT:PRINT
830 PRINT"TO DO ANOTHER SET, PRESS ANY KEY"
840 GET KY$
850 CLEAR:GOTO 10

```

Sample Run

```

*****
* INVENTORY REVIEW *
*****

```

```

MANUFACTURER:
BROWN      RETURN
PRODUCT:
WIDGET     RETURN
RETAIL PRICE: $
9.95       RETURN
QUANTITY ON HAND:
100        RETURN

```

```

MANUFACTURER:
SMITH      RETURN
PRODUCT:
GLOVES     RETURN

```

RETAIL PRICE: \$
12.99 RETURN
QUANTITY ON HAND:
40 RETURN

MANUFACTURER:
JONES RETURN
PRODUCT:
OVENS RETURN
RETAIL PRICE: \$
39.99 RETURN
QUANTITY ON HAND:
17 RETURN

MANUFACTURER:
RETURN

SORTING

MFG / ITEM / PRICE / QUANTITY

BROWN WIDGET \$9.95 100
JONES OVENS \$39.99 17
SMITH GLOVES \$12.99 40

TO REVIEW INVENTORY, PRESS ANY KEY
H

INVENTORY REVIEW
NUMBER DIFFERENT ITEMS:
Ø FROM \$40.00 ON UP
1 FROM \$30.00 TO 39.99
Ø FROM \$20.00 TO 29.99
1 FROM \$10.00 TO 19.99
1 UNDER \$10.00

TOTAL 3 ITEMS

AVERAGE PRICE \$20.9766667
RANGE \$9.95 TO \$39.99
TOTAL RETAIL VALUE \$2194.43
TO DO ANOTHER SET,PRESS ANY KEY

Invoice Computer

The computer will ask you for a discount percentage to be applied to the invoice; the retail price of goods being invoiced; and the quantity of those goods.

You enter actual percentage. The computer changes that to the appropriate decimal value.

Then it will ask if you have other items to be shown on the same invoice. If so, it again will get price and quantity sold info from you. It will assume the same discount applies.

When you tell the computer there are no more items for the same invoice, it will compute and display the total.

Program Listing

```
10 GOSUB 600
20 PRINT:PRINT"INFORMATION FOR
   COMPUTING INVOICES:"
30 PRINT:INPUT"WHAT IS DISCOUNT
   PERCENTAGE ";DP
40 DD=1-.01*DP
50 INPUT"WHAT IS THE ITEM SOLD ";IT$
60 INPUT"WHAT IS RETAIL PRICE ";P
70 INPUT"WHAT IS THE QUANTITY SOLD ";Q
80 C=P*DD:I=C*Q:T=T+I
90 PRINT:PRINT"WHOLESALE COST OF ";Q;
   " ";IT$;"S: $";I
100 PRINT:PRINT:PRINT
110 PRINT"IF THERE ARE MORE ITEMS"
120 PRINT"FOR THE SAME INVOICE, PRESS M"
130 PRINT:PRINT"IF NO OTHER ITEMS,PRESS
140 Z=PEEK(-16384)
150 POKE -16368,0
160 IF Z<128 THEN 140
170 IF Z=205 THEN HOME:GOTO 50
180 IF Z=212 THEN 200
190 GOTO 140
200 PRINT:PRINT CHR$(7)
210 PRINT"INVOICE GRAND TOTAL: $";T
220 PRINT:PRINT
230 PRINT"TO DO OTHER INVOICES"
```

```

240 PRINT"AT OTHER DISCOUNTS, PRESS
    ANY KEY"
250 GET KY$
260 CLEAR:GOTO 10
600 HOME
610 PRINT "*****"
620 PRINT "* INVOICE COMPUTER *"
630 PRINT "*****"
640 RETURN

```

Pricing Mark-Up

Figuring price mark-ups can be boring work. The computer can lighten that load.

This program works for manufacturers, wholesalers, and retailers. For instance, a manufacturing firm, selling products to middlemen or to the public, might calculate wholesale or retail price using a standard pricing *multiplier*. That number, multiplied times manufacturing cost, gives price. Such a standard multiplier is used by companies where exact manufacturing and production costs are too complex to calculate precisely.

Suppose a manufacturing firm uses an 8x multiplier. It multiplies manufacturing/production costs by 8 to set price. If one product has a manufacturing cost of \$1.89, the 8x formula would show a price of \$15.12.

Many firms like to round up to the next 95¢. In this example, the price would become \$15.95.

Lines 300 to 590 of the program handle such wholesale price-multiplier problems for you.

A retail store knows the discount it gets from a supplier. The store has the supplier's invoice showing wholesale price. This program will compute the list price from the discount and wholesale price.

For instance, suppose your wholesaler gives you a 42 percent discount on a wholesale \$49 item. The computer finds that list retail is \$84.48.

Lines 600 to 1120 of our program handle such retail pricing for you.

Lines 10 to 70 create a program-opening billboard.

Lines 100 to 200 allow you to select wholesale vs. retail.
Lines 1200-1230 create a closing billboard so you know
for sure the program is done running.

Program Listing

```
10 HOME: CLEAR: PRINT CHR$(7)
20 M$="PRICING MARK-UP"
30 LM=LEN(M$)
40 LT=LM+4
50 FOR L=1 TO LT
60 AS$=AS$+"*"
70 NEXT L
80 P=(40-LT)/2
90 HOME
100 PRINT TAB(P) AS$
110 PRINT TAB(P) "*" ";M$;" "*"
120 PRINT TAB(P) AS$
130 FOR L=1 TO 9:PRINT:NEXT L
140 PRINT TAB(9)"PRESS ANY KEY TO START"
145 FOR L=1 TO 9:PRINT:NEXT L
150 GET KY$
160 HOME:PRINT CHR$(7)
170 PRINT"WHICH KIND DO YOU WANT"
180 PRINT"RETAIL OR WHOLESALE ?"
190 PRINT:PRINT"RETAIL","PRESS R
    AND RETURN"
200 PRINT"WHOLESALE","PRESS W AND RETURN"
210 INPUT CH$
220 IF CH$="W" THEN 300
230 IF CH$="R" THEN 600
240 GOTO 210
300 HOME:PRINT CHR$(7)
310 INPUT"MANUFACTURING COST: $";MC
320 INPUT"PRICING MULTIPLIER: ";PM
330 WP=MC*PM
340 HOME:PRINT CHR$(7)
350 PRINT"DO YOU WANT EXACT PRICE,
    NOT ROUNDED"
360 PRINT"OR PRICE TO NEAREST 95 CENTS ?"
370 PRINT:PRINT"EXACT","PRESS E AND
    RETURN"
```



```

380 PRINT"NEAREST 95","PRESS N AND
    RETURN"
390 INPUT A$
400 IF A$="E" THEN 490
410 IF A$="N" THEN 440
420 GOTO 390
440 WP=INT(MC*PM)+.95
490 HOME:PRINT CHR$(7)
500 B$="COST: ";C$="MULTIPLIER
    : ";D$="WHOLESALE: "
510 PRINT B$+"$";MC
520 PRINT C$;PM
530 PRINT D$+"$";WP
540 FOR L=1 TO 8:PRINT:NEXT L
550 PRINT"TO DO MORE","PRESS M AND RETURN"
560 PRINT"TO QUIT","PRESS Q AND RETURN"
570 INPUT A$
580 IF A$="M" THEN CLEAR:GOTO 160
590 GOTO 1200
600 HOME:PRINT CHR$(7)
610 PRINT"DO YOU WANT EXACT CENTS"
620 PRINT"OR NEAREST 95 CENTS ?"
630 PRINT:PRINT"EXACT","PRESS E AND
    RETURN"
640 PRINT"NEAREST 95","PRESS N AND RETURN"
650 INPUT A$
660 IF A$="E" THEN 800
670 IF A$="N" THEN 1000
680 GOTO 650
800 HOME:PRINT CHR$(7)
810 INPUT"WHOLESALE PRICE YOU PAID
    : $";WP
820 INPUT"DISCOUNT PERCENT YOU WERE GIVEN
    : ";WD
830 PC=.01*WD:RP=WP/(1-PC)
840 HOME:PRINT CHR$(7)
850 PRINT"WHOLESALE: $";WP
860 PRINT"DISCOUNT: ";WD;"%"
870 PRINT"RETAIL: $";RP
880 FOR L=1 TO 9:PRINT:NEXT L
890 PRINT"TO DO MORE","PRESS M AND
    RETURN"

```

```

900 PRINT"TO QUIT","PRESS Q AND RETURN"
910 PRINT:PRINT:INPUT A$
920 IF A$="M" THEN CLEAR:GOTO 160
930 GOTO 1200
1000 HOME:PRINT CHR$(7)
1010 INPUT"WHOLESALE PRICE YOU PAID
      : $";WP
1020 INPUT"MARK-UP PERCENT YOU WANT
      : ";MU
1030 RP=INT(WP/(1-.01*MU))+.95
1040 HOME:PRINT CHR$(7)
1050 PRINT"WHOLESALE: $";WP
1060 PRINT"MARK-UP: ";MU;"%"
1070 PRINT"RETAIL: $";RP
1080 FOR L=1 TO 9:PRINT:NEXT L
1090 PRINT"TO DO MORE","PRESS M AND
      RETURN"
1100 PRINT"TO QUIT","PRESS Q AND RETURN"
1110 PRINT:PRINT:INPUT A$
1120 IF A$="M" THEN CLEAR:GOTO 160
1200 HOME:PRINT CHR$(7)
1210 FOR L=1 TO 12:PRINT:NEXT L
1220 PRINT TAB(15)"THANK YOU"
1230 END

```

Sample Run

```

PRICING MARK-UP
PRESS ANY KEY TO START
H

```

```

WHICH KIND DO YOU WANT
RETAIL OR WHOLESALE ?

```

```

RETAIL      PRESS R AND RETURN
WHOLESALE   PRESS W AND RETURN
R           RETURN

```

```

DO YOU WANT EXACT CENTS
OR NEAREST 95 CENTS ?

```

```

EXACT PRESS E AND RETURN

```

NEAREST 95 PRESS N AND RETURN
N RETURN

WHOLESALE PRICE YOU PAID: \$
6.00 RETURN
MARK-UP PERCENT YOU WANT:
40 RETURN

WHOLESALE: \$6
MARK-UP: 40%
RETAIL: \$10.95

TO DO MORE PRESS M AND RETURN
TO QUIT PRESS Q AND RETURN

Daily Codes

Businesses everywhere are concerned about security. Banks, credit card managers, warehousemen, shipping clerks, office managers, all need private daily codes for internal use to prevent unauthorized admission to storage areas, financial records, private files.

Now you can use your own Apple computer to generate a set of secret codes, one for each day of the week. This program generates a series of pseudorandom numbers and displays a table of those numbers alongside names of the days of the week.

The subroutine in lines 200 to 220 generates four-digit random numbers.

Program Listing

```
10 HOME
20 GOSUB 200
100 PRINT "SUNDAY: ";C:GOSUB 200
110 PRINT "MONDAY: ";C:GOSUB 200
120 PRINT "TUESDAY: ";C:GOSUB 200
130 PRINT "WEDNESDAY: ";C:GOSUB 200
140 PRINT "THURSDAY: ";C:GOSUB 200
150 PRINT "FRIDAY: ";C:GOSUB 200
160 PRINT "SATURDAY: ";C:END
```

```

200 C=INT(10000*(RND(1)))
210 IF C<1000 THEN GOTO 200
220 RETURN

```

Hourly Wages

This handy program computes total hours worked at regular pay and number of hours worked at time-and-a-half overtime. It then finds gross pay and rounds off to the nearest cent. The program knows that overtime starts after 40 hours.

The result is a nice chart on the display, after the computer asks only three questions.

You can change the number of regular-work hours per week by changing the number 40 in lines 200 and 210. Change the overtime multiplier of 1.5 by changing the number 1.5 in line 300.

This program makes payroll booking quick and simple.

Program Listing

```

10 HOME:PRINT CHR$(7)
20 PRINT:PRINT
30 PRINT"*****"
40 PRINT"* HOURLY WAGES *"
50 PRINT"*****"
60 PRINT:PRINT
100 INPUT"EMPLOYEE'S NAME: ";N
110 PRINT CHR$(7)
120 INPUT"HOURLY PAY RATE: ";P
130 PRINT CHR$(7)
140 INPUT"TOTAL NUMBER OF HOURS WORKED
    : ";H
150 PRINT CHR$(7)
200 IF H=40 OR H<40 THEN RH=H:OH=0
    :GOTO 300
210 IF H>40 THEN OH=H-40:RH=40
300 OP=1.5*P:PA=RH*P:PB=OH*OP
310 PY=PA+PB
320 PY=INT(100*PY+.5)/100
400 HOME

```

```

410 PRINT"EMPLOYEE:",N$
420 PRINT
430 PRINT"TOTAL HOURS:",H
440 PRINT
450 PRINT"REGULAR HOURS:",RH
460 PRINT"REGULAR RATE:", "$";P
470 PRINT"TOTAL REGULAR:", "$";PA
480 PRINT
490 PRINT"OVERTIME HOURS:",OH
500 PRINT"OVERTIME RATE:", "$";OP
510 PRINT"TOTAL OVERTIME:", "$";PB
520 PRINT
530 PRINT"GROSS PAY:", "$";PY
600 FOR L=1 TO 6:PRINT:NEXT L
610 PRINT"TO COMPUTE WAGES FOR A"
620 PRINT"DIFFERENT EMPLOYEE, PRESS ANY KEY"
630 GET KY$
640 GOTO 10

```

Sample Run

```

*****
* HOURLY WAGES *
*****

```

```

EMPLOYEE'S NAME:
JONES          RETURN
HOURLY PAY RATE:
5.00          RETURN
TOTAL NUMBER OF HOURS WORKED:
50            RETURN

```

EMPLOYEE: JONES

TOTAL HOURS: 50

REGULAR HOURS: 40
 REGULAR RATE: \$5
 TOTAL REGULAR: \$200

OVERTIME HOURS: 10

OVERTIME RATE: \$7.5
TOTAL OVERTIME: \$75

GROSS PAY: \$2.75

TO COMPUTE WAGES FOR A
DIFFERENT EMPLOYEE, PRESS ANY KEY

Highest/Lowest Item Price

Got a group of products with a wide range of list prices? Use this program to sort them and you'll find which is highest priced and which is lowest.

Where the program asks for a "name," key in the product name. When it asks for "price," key in the list price.

By the way, we've set this up for 100 products. At line 20, we dimensioned the memory for 100. You can change that number to any lower or higher figure which suits your need.

To make the computer display the results of its computations, simply press RETURN without data in response to the "name" question. That will prompt the computer to go to lines 140 to 180, to show the highest and lowest prices.

Program Listing

```
10 HOME
20 DIM M$(100)
30 FOR L=1 TO 100
40 INPUT"NAME: ";MN$
50 IF MN$="" THEN 140
60 INPUT"PRICE: $";KW$
70 M$(L)=MN$+" "+KW$
80 N=N+1
90 K=VAL(KW$)
100 IF N=1 THEN LL=K:LM$=M$(L):HH=K
    :HM$=M$(L)
110 IF K<LL THEN LL=K:LM$=M$(L)
```

```

120 IF K>HH THEN HH=K:HM$=M$(L)
130 NEXT L
140 PRINT:PRINT CHR$(7)
150 PRINT"LOWEST PRICED ITEM: ";LM$
160 PRINT"HIGHEST PRICED ITEM: ";HM$
170 PRINT:PRINT:PRINT
180 PRINT"TO DO MORE, PRESS ANY KEY"
190 GET KY$
200 CLEAR:GOTO 10

```

Sample Run

```

NAME:
HAT    RETURN
PRICE: $
9.95   RETURN
NAME:
SCARF  RETURN
PRICE: $
2.99   RETURN
NAME:
GLOVES RETURN
PRICE: $
6.25   RETURN
NAME:
      RETURN

```

```

LOWEST PRICED ITEM: SCARF 2.99
HIGHEST PRICED ITEM: HAT 9.95

```

TO DO MORE, PRESS ANY KEY

Salesman's Commission

Representatives, salesmen, account representatives, sales representatives. Here's the no-sweat way to compute commissions to be paid to your sales corps.

The computer will ask you for pertinent data and then display results including the salesman's name, the pay

period, his commission percentage rate, gross sales, and commission payable.

Program Listing

```
10 HOME:PRINT CHR$(7)
20 PRINT TAB(12)"REP COMMISSIONS"
30 PRINT:PRINT
40 INPUT"SALES PERIOD ENDING DATE
   : ";D$
50 INPUT"SALESMAN'S NAME: ";N$
60 INPUT"COMMISSION PERCENTAGE: ";P
70 K=.01*P
80 INPUT"SALESMAN'S GROSS SALES: $";Q
90 T=K*Q
100 HOME:PRINT CHR$(7)
110 PRINT"SALESMAN:",N$
120 PRINT"PERIOD ENDING:",D$
130 PRINT"COMMISSION RATE:",P;"%"
140 PRINT"GROSS SALES:", "$";Q
150 PRINT"COMMISSION:", "$";T
160 FOR L=1 TO 9:PRINT:NEXT L
170 PRINT "FOR ANOTHER, PRESS ANY KEY"
180 GET KY$
190 P=0:K=0:T=0:PRINT CHR$(7)
200 HOME:GOTO 50
```

Sample Run

REP COMMISSIONS

SALES PERIOD ENDING DATE:

12/31/83 RETURN

SALESMAN'S NAME:

SMITH RETURN

COMMISSION PERCENTAGE:

15 RETURN

SALESMAN'S GROSS SALES: \$

16243.00 RETURN

SALESMAN: SMITH
PERIOD ENDING: 12/31/83
COMMISSN RATE: 15%
GROSS SALES: \$16243
COMMISSION: \$2436.45

FOR ANOTHER, PRESS ANY KEY

Employee Roll Sorter

Here's a simple sorting routine which you can use to keep your list of employees in order. Type the names of all your employees into the computer and it will use this "bubble sort" program to put the list in alphabetical order for you. As set up, it accepts up to 100 names.

Push RETURN (with no name entered) to get out of the input cycle.

Program Listing

```
10 HOME
20 DIM M$(100)
30 FOR L=1 TO 100
40 INPUT"EMPLOYEE'S NAME: ";N$
50 IF N$="" THEN 100
60 INPUT"EMPLOYEE DATA: ";D$
70 M$(L)=N$+" "+D$
80 PRINT
90 NEXT L
100 HOME:PRINT"SORTING"
110 T=0
120 FOR L=1 TO 99
130 IF M$(L)<=M$(L+1) THEN 150
140 E$=M$(L):M$(L)=M$(L+1):M$(L+1)=E$
    :T=1
150 NEXT L
160 IF T=1 THEN 110
170 HOME:PRINT CHR$(7)
200 FOR L=1 TO 100
210 IF M$(L)<>"" THEN PRINT M$(L)
220 NEXT L
```

Sample Run

```
EMPLOYEE'S NAME
JONES      RETURN
EMPLOYEE DATA
83127      RETURN
EMPLOYEE'S NAME
SMITH      RETURN
EMPLOYEE DATA
26549      RETURN
EMPLOYEE'S NAME
ANDREWS    RETURN
EMPLOYEE DATA
50719      RETURN
EMPLOYEE'S NAME
          RETURN
```

SORTING

```
ANDREWS 50719
JONES 83127
SMITH 26549
```

Unit Price

Suppose you find 895 green Widgets and buy them for \$695. How much did each green Widget cost? Rounded off, \$7.77.

Unit price is total price divided by quantity. The quantity can be expressed in weight, total numbers, etc. It works the same whether you are talking about pounds of coffee, yards of concrete, gallons of ice cream, boxes of books, or units of Widgets.

This program asks for the name of the item, quantity purchased and total price paid. It then displays quantity, name, total and unit price.

Program Listing

```
10 HOME
20 PRINT "*****"
```

```

30 PRINT"* UNIT PRICE *"
40 PRINT"*****"
50 PRINT:PRINT
100 INPUT"NAME OF ITEM ";N$
110 INPUT"QUANTITY OF ITEMS ";Q
120 INPUT"TOTAL PRICE FOR ALL ITEMS $";P
130 U=P/Q:PRINT CHR$(7)
200 PRINT:PRINT Q;" ";N$+"S";" TOTAL $";P
210 PRINT"EACH ";N$;" $";U
220 FOR L=1 TO 9:PRINT:NEXT L
230 PRINT"FOR ANOTHER,PRESS ANY KEY"
240 GET KY$
250 CLEAR:GOTO 20

```

Sample Run

UNIT PRICE

NAME OF ITEM
WIDGET RETURN

QUANTITY OF ITEMS
150 RETURN

TOTAL PRICE FOR ALL ITEMS \$
88.75 RETURN

150 WIDGETS TOTAL \$88.75
EACH WIDGET \$.591666667

FOR ANOTHER,PRESS ANY KEY

Executive Decision Maker

Stumped by a toughie? Got one too hot to handle alone? Need help with major decisions? When there is no other way to decide, punch up this quickie and get a definite YES or NO.

Program Listing

```
10 HOME:R=INT(1000*(RND(1)))
20 IF R>499 THEN PRINT "YES":GOTO 40
30 PRINT"NO"
40 GET KY$:PRINT CHR$(7):GOTO 10
```

Programs for Teachers and Students

Math Flasher

Here's the basic routine (no pun intended) for an educational flash-card program. This one is bare-bones, no frills. You can dress it up with more colorful right-n-wrong messages, opening and closing billboards, etc. You could even make it keep score and present a "batting average" at the end of its run.

Here's how it works:

Lines 10- 90 determine which type of math you wish to do. Lines 50- 80 move program action to the appropriate group of lines further along in the program.

Lines 200-280 handle addition. Lines 300-390, subtraction. Lines 400-480, multiplication. Lines 500-590, division.

For example, look at lines 200-280. Two separate random numbers are generated (lines 200 and 210). The random numbers are labeled P and Q. At line 220, the program uses P and Q and asks you to add them together. Line 230 waits for and accepts your answer.

Line 240 clears old, unwanted words from the screen to reduce confusion. At line 250, the program makes the right or wrong decision, using the powerful IF/THEN statement. Line 270 prints the correct answer.

Program execution for subtract (lines 300-390), multiply (lines 400-480), and divide (lines 500-590), are similar except for line 320 in subtraction and line 520 in division.

We make the assumption that it is not desirable to have negative numbers as results of subtraction. That is, we want only subtraction problems with results of zero, one, two, three, or higher. We want no problems which would result in answers below zero such as -1, -2, -3, and so forth. So, line 320 tests P and Q, before presenting the problem on the screen. If they will result in a negative-number answer, then the program returns to lines 300-310 for two new numbers.

In division, we want whole-number answers. That is, we want answers like 2 or 11 or 26. Not answers like 1.81 or 9.75 or 21.3343. So, line 520 tests P and Q to make sure

their dividend will be a whole number. If not, the program goes back to line 500 and line 510 for two new numbers.

Program Listing

```
10 HOME:PRINT"DO YOU WANT TO"
20 PRINT"ADD":PRINT"SUBTRACT"
30 PRINT"MULTIPLY":PRINT"DIVIDE"
40 PRINT:INPUT"WHICH ?";B$
50 IF B$="ADD" THEN GOTO 200
60 IF B$="SUBTRACT" THEN GOTO 300
70 IF B$="MULTIPLY" THEN GOTO 400
80 IF B$="DIVIDE" THEN GOTO 500
90 GOTO 40
200 P=INT(10*(RND(1)))
210 Q=INT(10*(RND(1)))
220 PRINT:PRINT"ADD ";P;" PLUS ";Q
230 INPUT R
240 HOME
250 IF R=P+Q THEN PRINT "CORRECT":GOTO 270
260 PRINT "WRONG"
270 PRINT P;" PLUS ";Q;" EQUALS ";P+Q
280 FOR LL=1 TO 1000:NEXT LL
290 PRINT:GOTO 200
300 P=INT(10*(RND(1)))
310 Q=INT(10*(RND(1)))
320 IF P-Q<0 THEN GOTO 300
330 PRINT:PRINT"SUBTRACT ";Q;" FROM ";P
340 INPUT R
350 HOME
360 IF R=P-Q THEN PRINT "CORRECT":GOTO 380
370 PRINT"WRONG"
380 PRINT P;" MINUS ";Q;" EQUALS ";P-Q
385 FOR LL=1 TO 1000:NEXT LL
390 PRINT:GOTO 300
400 P=INT(10*(RND(1)))
410 Q=INT(10*(RND(1)))
420 PRINT:PRINT"MULTIPLY ";P;" TIMES "
    ;Q
430 INPUT R
440 HOME
```

```

450 IF R=P*Q THEN PRINT "CORRECT"
      :GOTO 470
460 PRINT"WRONG"
470 PRINT P;" TIMES ";Q;" EQUALS ";P*Q
480 FOR LL=1 TO 1000:NEXT LL
490 PRINT:GOTO 400
500 P=INT(100*(RND(1)))
510 Q=INT(10*(RND(1)))
515 IF Q<1 THEN GOTO 510
520 IF P/Q <> INT(P/Q) THEN GOTO 500
530 PRINT "DIVIDE ";P;" BY ";Q
540 INPUT R
550 HOME
560 IF R=P/Q THEN PRINT "CORRECT"
      :GOTO 580
570 PRINT "WRONG"
580 PRINT P;" DIVIDED BY ";Q;" EQUALS "
      ;P/Q
585 FOR LL=1 TO 1000:NEXT LL
590 PRINT:GOTO 500

```

Exam Score Sorting

The final number scores of a large number of test results can be categorized and thereby cut down into a smaller quantity of numbers easily.

This program accepts exam scores and divides them into ranges we have labeled A, B, C, D and F. The program looks for test scores in a range of zero to 100. The predetermined grade ranges are F=0 to 59; D=60 to 69; C=70 to 79; B=80 to 89; and A=90 to 100.

You key in the letter X to break the entry cycle. Lines 100 to 140 sort the scores into letter grades A through F. Lines 150 to 170 sort highest and lowest scores. Line 200 finds the mid-range and average scores.

Program Listing

```

10 HOME: CLEAR
20 PRINT CHR$(7)
30 PRINT "ENTER A GROUP OF SCORES"

```



```

40 PRINT "FROM ZERO TO 100, ONE AT A
    TIME"
50 PRINT "ENTER X AFTER LAST SCORE"
60 PRINT: INPUT "SCORE = "; G$
70 IF G$="X" THEN 200
80 G=VAL(G$)
90 N=N+1
100 IF G<60 THEN F=F+1:GOTO 150
110 IF G<70 THEN D=D+1:GOTO 150
120 IF G<80 THEN C=C+1:GOTO 150
130 IF G<90 THEN B=B+1:GOTO 150
140 A=A+1
150 IF N=1 THEN L=G:H=G
160 IF G<L THEN L=G
170 IF G>H THEN H=G
180 S=S+G
190 GOTO 60
200 P=S/N:M=L+((H-L)/2)
210 HOME:PRINT"THERE WAS A TOTAL OF"
    ;N;" SCORES"
220 PRINT"RANGING FROM ";L;" TO ";H
230 PRINT"MID-RANGE SCORE WAS ";M
240 PRINT"AVERAGE SCORE WAS ";P
250 PRINT:PRINT"TOTALS FOR EACH LETTER
    GRADE:"
260 PRINT"A: ";A
270 PRINT"B: ";B
280 PRINT"C: ";C
290 PRINT"D: ";D
300 PRINT"F: ";F
310 PRINT: CLEAR:GOTO 20

```

U.S. Presidents

Fourteenth. Let's see, that was Franklin Pierce. *Correct.* The fourteenth president was Franklin Pierce. Let's try another. *Thirty-fourth.* John F. Kennedy. *Wrong.* The thirty-fourth president was Dwight D. Eisenhower.

How many of the 40 U.S. presidents can you name?
Bet not as many as you would like!

This program tests not only your knowledge of the name of each president and his number in rank, but also the spelling of his name.

The more you take this test, the more you learn.

Program Listing

```
10 HOME
80 PRINT"HOW MANY U.S. PRESIDENTS
   CAN YOU NAME"
90 R=INT(80*(RND(1)))
100 IF R<1 THEN 90
110 IF INT(R/2)=R/2 THEN R=R+1
120 FOR L=1 TO R
130 READ S$
140 NEXT L
150 PRINT:PRINT
160 PRINT"WHO WAS THE "
170 PRINT S$
180 PRINT"PRESIDENT OF THE UNITED
   STATES"
190 READ C$
200 INPUT D$
210 PRINT
220 IF D$=C$ THEN PRINT "CORRECT"
   :GOTO 240
230 PRINT "WRONG"
240 PRINT "THE ";S$;" PRESIDENT WAS ";C$
250 RESTORE
260 PRINT:PRINT
270 GOTO 90
320 DATA FIRST,GEORGE WASHINGTON
330 DATA SECOND,JOHN ADAMS
340 DATA THIRD,THOMAS JEFFERSON
350 DATA FOURTH,JAMES MADISON
360 DATA FIFTH,JAMES MONROE
370 DATA SIXTH,JOHN QUINCY ADAMS
380 DATA SEVENTH,ANDREW JACKSON
390 DATA EIGHTH,MARTIN VAN BUREN
400 DATA NINTH,WILLIAM H. HARRISON
410 DATA TENTH,JOHN TYLER
420 DATA ELEVENTH,JAMES K. POLK
```

430 DATA TWELFTH,ZACHARY TAYLOR
440 DATA THIRTEENTH,MILLARD FILLMORE
450 DATA FOURTEENTH,FRANKLIN PIERCE
460 DATA FIFTEENTH,JAMES BUCHANAN
470 DATA SIXTEENTH,ABRAHAM LINCOLN
480 DATA SEVENTEENTH,ANDREW JOHNSON
490 DATA EIGHTEENTH,ULYSSES S. GRANT
500 DATA NINETEENTH,RUTHERFORD B. HAYES
510 DATA TWENTIETH,JAMES A. GARFIELD
520 DATA TWENTY-FIRST,CHESTER A. ARTHUR
530 DATA TWENTY-SECOND,GROVER CLEVELAND
540 DATA TWENTY-THIRD,BENJAMIN HARRISON
550 DATA TWENTY-FOURTH,GROVER CLEVELAND
560 DATA TWENTY-FIFTH,WILLIAM MCKINLEY
570 DATA TWENTY-SIXTH,THEODORE ROOSEVELT
580 DATA TWENTY-SEVENTH,WILLIAM H. TAFT
590 DATA TWENTY-EIGHTH,WOODROW WILSON
600 DATA TWENTY-NINTH,WARREN G. HARDING
610 DATA THIRTIETH,CALVIN COOLIDGE
620 DATA THIRTY-FIRST,HERBERT HOOVER
630 DATA THIRTY-SECOND,FRANKLIN D. ROOSEVELT
640 DATA THIRTY-THIRD,HARRY S. TRUMAN
650 DATA THIRTY-FOURTH,DWIGHT D. EISENHOWER
660 DATA THIRTY-FIFTH,JOHN F. KENNEDY
670 DATA THIRTY-SIXTH,LYNDON B. JOHNSON
680 DATA THIRTY-SEVENTH,RICHARD M. NIXON
690 DATA THIRTY-EIGHTH,GERALD R. FORD
700 DATA THIRTY-NINTH,JIMMY CARTER
710 DATA FOURTIETH,RONALD REAGAN

Number Of Days In A Month

Here's a cute teacher for your elementary-age kids. This program displays the name of a month and asks how many days in that month. If the correct number of days is entered, the computer says "correct." If an incorrect number of days is entered, the computer says "wrong." In either case, the correct answer is displayed. The educational game can go on forever if needed.

Program Listing

```
10 HOME
20 DATA JANUARY,31
30 DATA FEBRUARY,28
40 DATA MARCH,31
50 DATA APRIL,30
60 DATA MAY,31
70 DATA JUNE,30
80 DATA JULY,31
90 DATA AUGUST,31
100 DATA SEPTEMBER,30
110 DATA OCTOBER,31
120 DATA NOVEMBER,30
130 DATA DECEMBER,31
140 R=INT(100*(RND(1)))
150 IF R>24 THEN GOTO 140
160 IF INT(R/2)=R/2 THEN R=R-1
170 FOR L=1 TO R
180 READ $
190 NEXT L
200 PRINT "MONTH IS ";S$
210 READ C$
220 INPUT "HOW MANY DAYS ";D$
230 IF D$=C$ THEN PRINT "CORRECT"
    :GOTO 300
240 PRINT "WRONG"
300 PRINT "NUMBER OF DAYS IS ";C$
310 RESTORE
320 PRINT
330 GOTO 140
```

Foreign Capitals

Here's a learning quiz we'll bet you haven't seen anywhere else. This program tests your knowledge of foreign countries. The more you play, the more you learn!

You must tell the computer the correct name of the capital of the country it presents. And you must spell the name of that city correctly.

What is the capital of Egypt, Poland, Turkey, New Zealand, Bolivia or Afghanistan? It can be very tough!

Want to change to different countries? Change the DATA lines 20 to 540. Be sure to put a comma between country and capital in each DATA line.

Program Listing

```
10 HOME
20 DATA AFGHANISTAN,KABUL
30 DATA ALBANIA,TIRANA
40 DATA ALGERIA,ALGIERS
50 DATA ARGENTINA,BUENOS AIRES
60 DATA AUSTRALIA,CANBERRA
70 DATA AUSTRIA,VIENNA
80 DATA BAHRAIN,MANAMA
90 DATA BANGLADESH,DACCA
100 DATA BELGIUM,BRUSSELS
110 DATA BOLIVIA,LA PAZ
120 DATA BRAZIL,BRASILIA
130 DATA BULGARIA,SOFIA
140 DATA BURMA,RANGOON
150 DATA CHILE,SANTIAGO
160 DATA COLOMBIA,BOGOTA
170 DATA CUBA,HAVANA
180 DATA CZECHOSLOVAKIA,PRAGUE
190 DATA DENMARK,COPENHAGEN
200 DATA EGYPT,CAIRO
210 DATA FINLAND,HELSINKI
220 DATA FRANCE,PARIS
230 DATA GERMANY EAST,EAST BERLIN
240 DATA GERMANY WEST,BONN
250 DATA GREECE,ATHENS
260 DATA HAITI,PORT-AU-PRINCE
270 DATA HUNGARY,BUDAPEST
280 DATA ICELAND,REYKJAVIK
290 DATA INDIA,NEW DELHI
300 DATA IRAN,TEHRAN
310 DATA ITALY,ROME
320 DATA JAPAN,TOKYO
```

```

330 DATA KUWAIT,KUWAIT
340 DATA LIBYA,TRIPOLI
350 DATA MEXICO,MEXICO CITY
360 DATA NEPAL,KATHMANDU
370 DATA NEW ZEALAND,WELLINGTON
380 DATA NORWAY,OSLO
390 DATA OMAN,MUSCAT
400 DATA PERU,LIMA
410 DATA POLAND,WARSAW
420 DATA QATAR,DOHA
430 DATA ROMANIA,BUCHAREST
440 DATA SPAIN,MADRID
450 DATA SUDAN,KHARTOUM
460 DATA SWEDEN,STOCKHOLM
470 DATA SWITZERLAND,BERN
480 DATA TURKEY,ANKARA
490 DATA U.S.S.R.,MOSCOW
500 DATA UNITED KINGDOM,LONDON
510 DATA VENEZUELA,CARACAS
520 DATA YUGOSLAVIA,BELGRADE
530 DATA ZAIRE,KINSHASA
540 DATA ZAMBIA,LUSAKA
550 PRINT"HOW MANY FOREIGN CAPITALS
    CAN YOU NAME ?"
560 R=INT(106*(RND(1)))
570 IF R<1 THEN 560
580 IF INT(R/2)=R/2 THEN R=R-1
590 FOR L=1 TO R
600 READ S$
610 NEXT L
620 PRINT:PRINT
630 PRINT"FOREIGN COUNTRY IS ";S$
640 READ C$
650 INPUT"WHAT IS THE CAPITAL ";D$
660 IF D$=C$ THEN PRINT"CORRECT"
    :GOTO 680
670 PRINT"WRONG"
680 PRINT"CAPITAL IS ";C$
690 RESTORE
700 PRINT:PRINT
710 GOTO 560

```

Weights & Measures

How many inches in a centimeter? Ounces in a tablespoon? Cubic feet of firewood in a cord? Run this program and you'll not only test your existing knowledge of weights and measurements but learn more.

To change the information in the quiz, change DATA in lines 20 to 510. Be sure to follow the wording format used here and put a comma between question and answer in each DATA line.

Program Listing

```
10 HOME
20 DATA MILE IN FEET,5280
30 DATA MILE IN KILOMETERS,1.609
40 DATA NAUTICAL MILE IN STATUTE
   MILES,1.151
50 DATA MILLIMETER IN INCHES,.03937
60 DATA POINT(TYPOGRAPHY) IN
   INCHES,.013837
70 DATA ROD IN YARDS,5.5
80 DATA ACRE IN SQUARE YARDS,4840
90 DATA ACRE IN SQUARE FEET,43560
100 DATA HECTARE IN ACRES,2.471
110 DATA CENTIMETER IN INCHES,0.3937
120 DATA SURVEYOR'S CHAIN IN FEET,66
130 DATA ENGINEER'S CHAIN IN FEET,100
140 DATA FATHOM IN FEET,6
150 DATA FOOT IN METERS,0.3048
160 DATA FURLONG IN FEET,660
170 DATA FURLONG IN YARDS,220
180 DATA FURLONG IN METERS,201.168
190 DATA HAND IN INCHES,4
200 DATA INCH IN CENTIMETERS,2.54
210 DATA KILOMETER IN MILES,0.621
220 DATA LEAGUE IN MILES,3
230 DATA METER IN INCHES,39.37
240 DATA METER IN YARDS,1.094
250 DATA SQ. FOOT IN SQ. CENTIMETERS,
   929.030
260 DATA SQ. METER IN SQ. YARDS,1.196
```

```

270 DATA SQUARE MILE IN HECTARES,258.999
280 DATA SQ. YARD IN SQUARE METERS,0.836
290 DATA BUSHEL IN CUBIC INCHES,2150.42
300 DATA CORD FIREWOOD IN CUBIC FEET,128
310 DATA CUBIC FOOT IN GALLONS,7.481
320 DATA CUP IN FLUID OUNCES,8
330 DATA CUP LIQUID IN PINTS,0.5
340 DATA U.S. GALLON IN LITERS,3.785
350 DATA LITER LIQUID IN QUARTS,1.057
360 DATA FLUID OUNCE IN MILLILITERS,29.574
370 DATA PECK IN LITERS,8.810
380 DATA DRY PINT IN CUBIC INCHES,33.6
390 DATA PINT LIQUID IN LITERS,0.473
400 DATA DRY QUART IN CUBIC INCHES,67.201
410 DATA QUART LIQUID IN LITERS,0.946
420 DATA TABLESPOON IN TEASPOONS,3
430 DATA TABLESPOON IN FLUID OUNCES,0.5
440 DATA TEASPOON IN TABLESPOONS,0.3333333
450 DATA GRAM IN OUNCES,0.035
460 DATA KILOGRAM IN POUNDS,2.205
470 DATA OUNCE IN GRAMS,28.350
480 DATA PENNYWEIGHT IN GRAMS,1.555
490 DATA POUND IN GRAMS,453.59237
500 DATA TON(NET OR SHORT) IN POUNDS,2000
510 DATA TON(METRIC) IN POUNDS,2204.623
520 PRINT"HOW MANY EQUIVALENTS OF"
530 PRINT"WEIGHTS AND MEASURES DO YOU
    KNOW ?"
540 R=INT(100*(RND(1)))
550 IF R<1 THEN 540
560 IF INT(R/2)=R/2 THEN R=R-1
570 FOR L=1 TO R
580 READ S$
590 NEXT L
600 PRINT:PRINT
610 PRINT"WHAT IS THE EQUIVALENT OF 1"
    ;S$
620 READ C$
630 INPUT D$
640 PRINT
650 IF D$=C$ THEN PRINT "CORRECT"
    :GOTO 670

```



```

660 PRINT "WRONG"
670 PRINT"THE CORRECT ANSWER IS ";C$
680 RESTORE
690 PRINT:PRINT
700 GOTO 540

```

State Geographic Centers

This mind bender tests your knowledge of geographic locations of cities and towns in the United States. These are special places since, in each case, they are the town nearest to the geographic center of its state.

In other words, Columbus happens to be almost exactly in the center of Ohio. But which state has Challis at its center? Or Lewistown? Or Oklahoma City? (Well, some may be obvious!)

You not only learn a lot from running this program but you have a barrel of fun. Talk about trivia!

Program Listing

```

10 HOME
20 DATA CLANTON,ALABAMA
30 DATA MT. MCKINLEY,ALASKA
40 DATA PRESCOTT,ARIZONA
50 DATA LITTLE ROCK,ARKANSAS
60 DATA MADERA,CALIFORNIA
70 DATA PIKES PEAK,COLORADO
80 DATA EAST BERLIN,CONNECTICUT
90 DATA DOVER,DELAWARE
100 DATA BROOKSVILLE,FLORIDA
110 DATA MACON,GEORGIA
120 DATA MAUI ISLAND,HAWAII
130 DATA CHALLIS,IDAHO
140 DATA SPRINGFIELD,ILLINOIS
150 DATA INDIANAPOLIS,INDIANA
160 DATA AMES,IOWA
170 DATA GREAT BEND,KANSAS
180 DATA LEBANON,KENTUCKY
190 DATA MARKSVILLE,LOUISIANA

```

```

200 DATA DOVER-FOXCROFT,MAINE
210 DATA DAVIDSONVILLE,MARYLAND
220 DATA WORCESTER,MASSACHUSETTS
230 DATA CADILLAC,MICHIGAN
240 DATA BRAINERD,MINNESOTA
250 DATA CARTHAGE,MISSISSIPPI
260 DATA JEFFERSON CITY,MISSOURI
270 DATA LEWISTOWN,MONTANA
280 DATA BROKEN BOW,NEBRASKA
290 DATA AUSTIN,NEVADA
300 DATA ASHLAND,NEW HAMPSHIRE
310 DATA TRENTON,NEW JERSEY
320 DATA WILLARD,NEW MEXICO
330 DATA ONEIDA,NEW YORK
340 DATA SANFORD,NORTH CAROLINA
350 DATA MCCLUSKY,NORTH DAKOTA
360 DATA COLUMBUS,OHIO
370 DATA OKLAHOMA CITY,OKLAHOMA
380 DATA PRINEVILLE,OREGON
390 DATA BELLEFONTE,PENNSYLVANIA
400 DATA CROMPTON,RHODE ISLAND
410 DATA COLUMBIA,SOUTH CAROLINA
420 DATA PIERRE,SOUTH DAKOTA
430 DATA MURFREESBORO,TENNESSEE
440 DATA BRADY,TEXAS
450 DATA MANTI,UTAH
460 DATA ROXBURY,VERMONT
470 DATA BUCKINGHAM,VIRGINIA
480 DATA WENATCHEE,WASHINGTON
490 DATA SUTTON,WEST VIRGINIA
500 DATA MARSHFIELD,WISCONSIN
510 DATA LANDER,WYOMING
520 PRINT"FOR HOW MANY STATES"
530 PRINT"CAN YOU NAME THE GEOGRAPHICAL
    CENTER ?"
540 R=INT(100*(RND(1)))
550 IF R<1 THEN 540
560 IF INT(R/2)=R/2 THEN R=R-1
570 FOR L=1 TO R
580 READ S$
590 NEXT L
600 PRINT:PRINT

```

```

610 PRINT"WHICH STATE HAS ITS"
620 PRINT"GEOGRAPHIC CENTER NEAR"
630 PRINT S$
640 READ C$
650 INPUT D$
660 IF D$=C$ THEN PRINT:PRINT"CORRECT"
    :GOTO 680
670 PRINT:PRINT"WRONG"
680 PRINT"THE STATE IS ";C$
690 RESTORE
700 PRINT:PRINT
710 GOTO 540

```

Alphabetizing Test Evaluation

Here's a more elaborate exam-score evaluation. As set here, you may enter scores for up to 50 students. The computer asks for last name, first name, and score number for each student.

The machine sorts students so it can present an alphabetized list at the end. As you will note in the sample run, the computer can tell the difference between people with the same last names, or with the same first names. If two or more students have the same last and the same first name, the computer sorts them by grade score.

If you are using a printer, you'll be able to make a permanent record of your fully-alphabetized list of names and scores.

You can change the total number of students in the group by changing the number 50 in lines 1010 and 1425 and the number 49 in line 1330.

Again we assume zero to 59, F; 60-69, D; 70-79, C; 80-89, B; and 90-100, A. You can change those ranges in lines 1110-1150.

Program Listing

```

1000 HOME: CLEAR
1010 DIM M$(50)
1020 FOR L=1 TO 50
1030 INPUT "LAST NAME: "; P$

```

```

1040 IF P$="" THEN 1300
1050 INPUT "FIRST NAME. ";Q$
1060 INPUT "GRADE: ";G$
1100 X=VAL(G$)
1110 IF X<60 THEN F=F+1:GOTO 1160
1120 IF X<70 THEN D=D+1:GOTO 1160
1130 IF X<80 THEN C=C+1:GOTO 1160
1140 IF X<90 THEN B=B+1:GOTO 1160
1150 A=A+1
1160 Y=Y+X
1170 Z=Z+1
1180 IF Z=1 THEN J=X:K=X
1190 IF X<J THEN J=X
1200 IF X>K THEN K=X
1210 I=I+X
1220 H=I/Z:MM=J+((K-J)/2)
1230 M$(L)=P$+" "+Q$+" "+G$
1240 NEXT L
1300 HOME
1310 PRINT "SORTING"
1320 T=0
1330 FOR L=1 TO 49
1340 IF M$(L)<=M$(L+1) THEN 1360
1350 E$=M$(L):M$(L)=M$(L+1):M$(L+1)=E$
      :T=1
1360 NEXT L
1370 IF T=1 THEN 1320
1400 HOME
1400 CLS
1410 PRINT "LAST NAME/FIRST NAME/GRADE"
1420 PRINT
1425 FOR L=1 TO 50
1430 IF M$(L)<>"" THEN PRINT M$(L)
1435 NEXT L
1440 PRINT
1450 PRINT"FOR STATISTICS, PRESS S"
1455 PRINT"TO QUIT, PRESS Q"
1460 GET AA$
1470 IF AA$="S" THEN 1500
1480 IF AA$="Q" THEN 1800
1490 GOTO 1460

```

```

1500 HOME
1510 PRINT "TEST SCORE STATISTICS"
1520 PRINT
1530 PRINT "SCORES RANGE ";J;" TO ";K
1540 PRINT "MID-RANGE SCORE IS ";MM
1550 PRINT "AVERAGE SCORE IS ";H
1560 PRINT "TOTALS FOR EACH LETTER GRADE:"
1570 PRINT A;" A"
1580 PRINT B;" B"
1590 PRINT C;" C"
1600 PRINT D;" D"
1610 PRINT F;" F"
1620 PRINT "A TOTAL OF ";Z;" SCORES"
1630 PRINT
1640 PRINT "TO DO ANOTHER SET, PRESS A"
1650 PRINT "TO STOP, PRESS S"
1660 GET BB$
1680 IF BB$="A" THEN 1000
1690 IF BB$="S" THEN 1800
1700 GOTO 1660
1800 HOME
1810 PRINT "                THANK YOU"
1820 GOTO 1820

```

Sample Run

```
RUN  RETURN
```

```

LAST NAME:
SMITH  RETURN
FIRST NAME:
ALAN  RETURN
GRADE:
87  RETURN

```

```

LAST NAME:
JONES  RETURN
FIRST NAME:
HELEN  RETURN
GRADE:
62  RETURN

```

LAST NAME:
FUDD RETURN
FIRST NAME:
ELMER RETURN
GRADE:
97 RETURN

LAST NAME:
MARTIN RETURN
FIRST NAME:
ALAN RETURN
GRADE:
79 RETURN

LAST NAME:
SMITH RETURN
FIRST NAME:
BARRY RETURN
GRADE:
55 RETURN

LAST NAME:
JONES RETURN
FIRST NAME:
HELEN RETURN
GRADE:
48 RETURN

LAST NAME:
RETURN
SORTING

LAST NAME/FIRST NAME/GRADE		
FUDD	ELMER	GRADE:97
JONES	HELEN	GRADE:48
JONES	HELEN	GRADE:62
MARTIN	ALAN	GRADE:79
SMITH	ALAN	GRADE:87
SMITH	BARRY	GRADE:55

FOR STATISTICS, PRESS S
TO QUIT, PRESS Q
S

TEST SCORE STATISTICS
SCORES RANGE 48 TO 97
MID-RANGE SCORE IS 72.5
AVERAGE SCORE IS 71.3333334

TOTALS FOR EACH LETTER GRADE:

1 A

1 B

1 C

1 D

2 F

A TOTAL OF 6 SCORES

TO DO ANOTHER SET, PRESS A

TO STOP, PRESS S

S

THANK YOU

Lakes of the World

Sure, Lake of the Woods and the Great Salt Lake are in North America. But where is Lake Van, Lake Onega, Lake Reindeer, or Lake Nyasa?

The computer asks you which continent is the location of a major large lake. To change the lakes, change DATA in lines 20 to 530. Be sure to put question before answer in each DATA line and separate with a comma.

Program Listing

```
10 HOME
20 DATA CASPIAN SEA,ASIA/EUROPE
30 DATA SUPERIOR,NORTH AMERICA
40 DATA VICTORIA,AFRICA
50 DATA ARAL SEA,ASIA
60 DATA HURON,NORTH AMERICA
70 DATA MICHIGAN,NORTH AMERICA
80 DATA TANGANYIKA,AFRICA
90 DATA BAYKAL,ASIA
100 DATA GREAT BEAR,NORTH AMERICA
```

110 DATA NYASA (MALAWI),AFRICA
120 DATA GREAT SLAVE,NORTH AMERICA
130 DATA CHAD,AFRICA
140 DATA ERIE,NORTH AMERICA
150 DATA WINNIPEG,NORTH AMERICA
160 DATA ONTARIO,NORTH AMERICA
170 DATA LADOGA,EUROPE
180 DATA BALKHASH,ASIA
190 DATA MARACAIBO,SOUTH AMERICA
200 DATA BANGWEULU,AFRICA
210 DATA TUNGTING,ASIA
220 DATA ONEGA,EUROPE
230 DATA EYRE,AUSTRALIA
240 DATA TITICACA,SOUTH AMERICA
250 DATA NICARAGUA,NORTH AMERICA
260 DATA ATHABASCA,NORTH AMERICA
270 DATA REINDEER,NORTH AMERICA
280 DATA TONLE SAP,ASIA
290 DATA TURKANA (RUDOLF),AFRICA
300 DATA ISSYK-KUL',ASIA
310 DATA URMIA,ASIA
320 DATA TORRENS,AUSTRALIA
330 DATA VANERN,EUROPE
340 DATA WINNIPEGOSIS,NORTH AMERICA
350 DATA ALBERT,AFRICA
360 DATA KARIBA,AFRICA
370 DATA NETTILLING,NORTH AMERICA
380 DATA CHANY,ASIA
390 DATA NIPIGON,NORTH AMERICA
400 DATA GAIRDNER,AUSTRALIA
410 DATA MWERU,AFRICA
420 DATA MANITOBA,NORTH AMERICA
430 DATA TAYMYR,ASIA
440 DATA KYOGA,AFRICA
450 DATA KHANKA,ASIA
460 DATA LAKE OF THE WOODS,NORTH AMERICA
470 DATA PEIPUS,EUROPE
480 DATA KOKO NOR,ASIA
490 DATA NASSER-NUBIA,AFRICA
500 DATA DUBAWNT,NORTH AMERICA
510 DATA VAN,ASIA
520 DATA WOLLASTON,NORTH AMERICA


```

530 DATA GREAT SALT LAKE,NORTH AMERICA
540 PRINT"HOW MANY FAMOUS LAKES OF THE
    WORLD CAN YOU LOCATE ?"
550 R=INT(104*(RND(1)))
560 IF R<1 THEN 550
570 IF INT(R/2)=R/2 THEN R=R-1
580 FOR L=1 TO R
590 READ S$
600 NEXT L
610 PRINT:PRINT
620 PRINT S$;" IS A LARGE LAKE"
630 READ C$
640 INPUT"ON WHAT CONTINENT OR CONTINENTS
    IS IT LOCATED";D$
650 IF D$=C$ THEN PRINT"CORRECT"
    :GOTO 670
660 PRINT"WRONG"
670 PRINT"THE CONTINENT IS ";C$
680 RESTORE
690 PRINT:PRINT
700 GOTO 550

```

Deserts of the World

If lakes are too wet for you, how about something more in the dry line? Deserts!

There are many major ones around the world. You may have heard of the Sahara or Death Valley. But where are they? On what continents? And where is Atacama, Gibson, Nefud or Sechura?

To change the items, modify DATA lines 20 to 280. Be sure to put a comma between question and answer in each DATA line.

Program Listing

```

10 HOME
20 DATA ARABIAN,AFRICA
30 DATA ATACAMA,SOUTH AMERICA
40 DATA COLORADO,NORTH AMERICA

```

```

50 DATA DASHT-I-KAVIR,ASIA
60 DATA DASHT-I-LUT,ASIA
70 DATA DEATH VALLEY,NORTH AMERICA
80 DATA GIBSON,AUSTRALIA
90 DATA GOBI,ASIA
100 DATA GREAT SALT LAKE,NORTH AMERICA
110 DATA GREAT SANDY,AUSTRALIA
120 DATA GREAT VICTORIA,AUSTRALIA
130 DATA KALAHARI,AFRICA
140 DATA KARA-KUM,ASIA
150 DATA KYZYL-KUM,ASIA
160 DATA MOJAVE,NORTH AMERICA
170 DATA NAMIB,AFRICA
180 DATA NEFUD (AN NAFUD),ASIA
190 DATA NEGEV,ASIA
200 DATA NUBIAN,AFRICA
210 DATA RUB AL KHALI,ASIA
220 DATA SAHARA,AFRICA
230 DATA SECHURA,SOUTH AMERICA
240 DATA SIMPSON,AUSTRALIA
250 DATA SYRIAN (EL HAMAD),ASIA
260 DATA TAKLAMAKAN,ASIA
270 DATA THAR (GREAT INDIAN),ASIA
280 DATA VIZCAINO,NORTH AMERICA
290 PRINT"HOW MANY DESERTS OF THE WORLD
    CAN YOU LOCATE ?"
300 R=INT(54*(RND(1)))
310 IF R<1 THEN 300
320 IF INT(R/2)=R/2 THEN R=R-1
330 FOR L=1 TO R
340 READ S$
350 NEXT L
360 PRINT:PRINT
370 PRINT S$
380 PRINT"IS ONE OF"
390 PRINT"WORLD'S GREAT DESERTS"
400 READ C$
410 PRINT:PRINT"ON WHAT CONTINENT"
420 INPUT"IS IT LOCATED";D$
430 IF D$=C$ THEN PRINT"CORRECT"
    :GOTO 450
440 PRINT"WRONG"

```

```
450 PRINT"CONTINENT IS ";C$
460 RESTORE
470 PRINT:PRINT
480 GOTO 300
```

Volcanoes of the World

In what country does the lava flow? Where is the active volcano we know as Mauna Loa? Lascar? Fogo? Torbert?

The computer names a volcano and you name the country, except in the one case in Antarctica which is a continent.

To change to other volcanoes, change DATA lines 20 to 280. Remember the commas.

Program Listing

```
10 HOME
20 DATA KILIMANJARO,TANZANIA
30 DATA EREBUS,ANTARCTICA
40 DATA KLYUCHEVSKAYA,USSR
50 DATA MAUNA KEA,USA
60 DATA ETNA,ITALY
70 DATA POPOCATEPETL,MEXICO
80 DATA GUALLATIRI,CHILE
90 DATA CAMEROON MT.,CAMEROON
100 DATA KERINTJI,INDONESIA
110 DATA MAUNA LOA,USA
120 DATA COLIMA,MEXICA
130 DATA TAJUMULCO,GUATEMALA
140 DATA LASCAR,CHILE
150 DATA FUJI,JAPAN
160 DATA WRANGELL,USA
170 DATA COTOPAXI,ECUADOR
180 DATA NYIRAGONGO,CONGO
190 DATA TOLBACHIK,USSR
200 DATA TORBERT,USA
210 DATA SPURR,USA
220 DATA LASSEN,USA
```

```

230 DATA TACANA,GUATEMALA
240 DATA IRAZU,COSTA RICA
250 DATA MISTI,PERU
260 DATA PURACE,COLOMBIA
270 DATA EL TEIDE,CANARY ISLANDS
280 DATA FOGO,CAPE VERDE ISLANDS
290 PRINT"HOW MANY VOLCANOES OF THE WORLD"
300 PRINT"CAN YOU LOCATE ?"
310 R=INT(54*(RND(1)))
320 IF R<1 THEN 310
330 IF INT(R/2)=R/2 THEN R=R-1
340 FOR L=1 TO R
350 READ S$
360 NEXT L
370 PRINT:PRINT
380 PRINT S$
390 PRINT"IS ONE OF THE"
400 PRINT"WORLD'S HIGHEST VOLCANOES"
410 READ C$
420 PRINT"IN WHAT COUNTRY"
430 INPUT"IS IT LOCATED";D$
440 IF D$=C$ THEN PRINT"CORRECT"
      :GOTO 460
450 PRINT:PRINT"WRONG"
460 PRINT"COUNTRY IS ";C$
470 RESTORE
480 PRINT:PRINT
490 GOTO 310

```

Chemistry:

Elements and their Symbols

Here's a great drill for the chemistry student. The computer displays one of the chemical symbols and you must reply with the correct name of the element. And it must be spelled correctly!

Do you recognize Au, Es, Pr, Tc, or Yb? Then you need this experience!

Program Listing

```
10 HOME
20 DATA AC,ACTINIUM
30 DATA AL,ALUMINUM
40 DATA AM,AMERICIUM
50 DATA SB,ANTIMONY
60 DATA AR,ARGON
70 DATA AS,ARSENIC
80 DATA AT,ASTATINE
90 DATA BA,BARIUM
100 DATA BK,BERKELIUM
110 DATA BE,BERYLLIUM
120 DATA BI,BISMUTH
130 DATA B,BORON
140 DATA BR,BROMINE
150 DATA CD,CADMIUM
160 DATA CA,CALCIUM
170 DATA CF,CALIFORNIUM
180 DATA C,CARBON
190 DATA CE,CERIUM
200 DATA CS,CESIUM
210 DATA CL,CHLORINE
220 DATA CR,CHROMIUM
230 DATA CO,COBALT
240 DATA CU,COPPER
250 DATA CM,CURIUM
260 DATA DY,DYSPROSIUM
270 DATA ES,EINSTEINIUM
280 DATA ER,ERBIUM
290 DATA EU,EUROPIUM
300 DATA FM,FERMIUM
310 DATA F,FLUORINE
320 DATA FR,FRANCIUM
330 DATA GD,GADOLINIUM
340 DATA GA,GALLIUM
350 DATA GE,GERMANIUM
360 DATA AU,GOLD
370 DATA HF,HAFNIUM
380 DATA HE,HELIUM
390 DATA HO,HOLMIUM
400 DATA H,HYDROGEN
```

410 DATA IN,INDIUM
420 DATA I,IODINE
430 DATA IR,IRIDIUM
440 DATA FE,IRON
450 DATA KR,KRYPTON
460 DATA LA,LANTHANUM
470 DATA LR,LAWRENCIUM
480 DATA PB,LEAD
490 DATA LI,LITHIUM
500 DATA LU,LUTETIUM
510 DATA MG,MAGNESIUM
520 DATA MN,MANGANESE
530 DATA MD,MENDELEVIUM
540 DATA HG,MERCURY
550 DATA MO,MOLYBDENUM
560 DATA ND,NEODYMIUM
570 DATA NE,NEON
580 DATA NP,NEPTUNIUM
590 DATA NI,NICKEL
600 DATA NB,NIOBIUM
610 DATA N,NITROGEN
620 DATA NO,NOBELIUM
630 DATA OS,OSMIUM
640 DATA O,OXYGEN
650 DATA PD,PALLADIUM
660 DATA P,PHOSPHORUS
670 DATA PT,PLATINUM
680 DATA PU,PLUTONIUM
690 DATA PO,POLONIUM
700 DATA K,POTASSIUM
710 DATA PR,PRASEODYMIUM
720 DATA PM,PROMETHIUM
730 DATA PA,PROTACTINIUM
740 DATA RA,RADIUM
750 DATA RN,RADON
760 DATA RE,RHENIUM
770 DATA RH,RHODIUM
780 DATA RB,RUBIDIUM
790 DATA RU,RUTHENIUM
800 DATA SM,SAMARIUM
810 DATA SC,SCANDIUM
820 DATA SE,SELENIUM

```

830 DATA SI,SILICON
840 DATA AG,SILVER
850 DATA NA,SODIUM
860 DATA SR,STRONTIUM
870 DATA S,SULFUR
880 DATA TA,TANTALUM
890 DATA TC,TECHNETIUM
900 DATA TE,TELLURIUM
910 DATA TB,TERBIUM
920 DATA TL,THALLIUM
930 DATA TH,THORIUM
940 DATA TM,THULIUM
950 DATA SN,TIN
960 DATA TI,TITANIUM
970 DATA W,TUNGSTEN
980 DATA U,URANIUM
990 DATA V,VANADIUM
1000 DATA XE,XENON
1010 DATA YB,YTTERBIUM
1020 DATA Y,YTTRIUM
1030 DATA ZN,ZINC
1040 DATA ZR,ZIRCONIUM
1050 PRINT"FOR HOW MANY CHEMICAL ELEMENTS"
1060 PRINT"DO YOU KNOW THE SYMBOLS ?"
1070 R=INT(206*(RND(1)))
1080 IF R<1 THEN 1070
1090 IF INT(R/2)=R/2 THEN R=R-1
1100 FOR L=1 TO R
1110 READ S$
1120 NEXT L
1130 PRINT:PRINT
1140 PRINT S$;" IS THE SYMBOL"
1150 PRINT "WHAT IS THE ELEMENT"
1160 READ C$
1170 INPUT D$
1180 PRINT
1190 IF D$=C$ THEN PRINT"CORRECT"
      :GOTO 1210
1200 PRINT"WRONG"
1210 PRINT"THE ELEMENT IS ";C$
1220 RESTORE
1230 PRINT:PRINT
1240 GOTO 1070

```

Chemistry:

Periodic Table of Elements

How many atomic numbers do you know? What is the atomic number of gold? Silver? Oxygen? Uranium? Neodymium? Ytterbium?

This program causes the computer to drill you mercilessly until you know every last one!

Program Listing

```
10 HOME
20 DATA ACTINIUM,89
30 DATA ALUMINUM,13
40 DATA AMERICIUM,95
50 DATA ANTIMONY,51
60 DATA ARGON,18
70 DATA ARSENIC,33
80 DATA ASTATINE,85
90 DATA BARIUM,56
100 DATA BERKELIUM,97
110 DATA BERYLLIUM,4
120 DATA BISMUTH,83
130 DATA BORON,5
140 DATA BROMINE,35
150 DATA CADMIUM,48
160 DATA CALCIUM,20
170 DATA CALIFORNIUM,98
180 DATA CARBON,6
190 DATA CERIUM,58
200 DATA CESIUM,55
210 DATA CHLORINE,17
220 DATA CHROMIUM,24
230 DATA COBALT,27
240 DATA COPPER,29
250 DATA CURIUM,96
260 DATA DYSPROSIUM,66
270 DATA EINSTEINIUM,99
280 DATA ERBIUM,68
290 DATA EUROPIUM,63
300 DATA FERMIUM,100
```


310 DATA FLUORINE,9
320 DATA FRANCIUM,87
330 DATA GADOLINIUM,64
340 DATA GALLIUM,31
350 DATA GERMANIUM,32
360 DATA GOLD,79
370 DATA HAFNIUM,2
380 DATA HELIUM,2
390 DATA HOLMIUM,67
400 DATA HYDROGEN,1
410 DATA INDIUM,49
420 DATA IODINE,53
430 DATA IRIDIUM,77
440 DATA IRON,26
450 DATA KRYPTON,36
460 DATA LANTHANUM,57
470 DATA LAWRENCIUM,103
480 DATA LEAD,82
490 DATA LITHIUM,3
500 DATA LUTETIUM,71
510 DATA MAGNESIUM,12
520 DATA MANGANESE,25
530 DATA MENDELEVIUM,101
540 DATA MERCURY,80
550 DATA MOLYBDENUM,42
560 DATA NEODYMIUM,60
570 DATA NEON,10
580 DATA NEPTUNIUM,93
590 DATA NICKEL,28
600 DATA NIOBIUM,41
610 DATA NITROGEN,7
620 DATA NOBELIUM,102
630 DATA OSMIUM,76
640 DATA OXYGEN,8
650 DATA PALLADIUM,46
660 DATA PHOSPHORUS,15
670 DATA PLATINUM,78
680 DATA PLUTONIUM,94
690 DATA POLONIUM,84
700 DATA POTASSIUM,19
710 DATA PRASEODYMIUM,59
720 DATA PROMETHIUM,61

```

730 DATA PROTACTINIUM,91
740 DATA RADIUM,88
750 DATA RADON,86
760 DATA RHENIUM,75
770 DATA RHODIUM,45
780 DATA RUBIDIUM,37
790 DATA RUTHENIUM,44
800 DATA SAMARIUM,62
810 DATA SCANDIUM,21
820 DATA SELENIUM,34
830 DATA SILICON,14
840 DATA SILVER,47
850 DATA SODIUM,11
860 DATA STRONTIUM,38
870 DATA SULFUR,16
880 DATA TANTALUM,73
890 DATA TECHNETIUM,43
900 DATA TELLURIUM,52
910 DATA TERBIUM,65
920 DATA THALLIUM,81
930 DATA THORIUM,90
940 DATA THULIUM,69
950 DATA TIN,50
960 DATA TITANIUM,22
970 DATA TUNGSTEN,74
980 DATA URANIUM,92
990 DATA VANADIUM,23
1000 DATA XENON,54
1010 DATA YTTERBIUM,70
1020 DATA YTTRIUM,39
1030 DATA ZINC,30
1040 DATA ZIRCONIUM,40
1050 PRINT"IN THE PERIODIC TABLE OF
      ELEMENTS"
1060 PRINT"HOW MANY ATOMIC NUMBERS DO
      YOU KNOW ?"
1070 R=INT(206*(RND(1)))
1080 IF R<1 THEN 1070
1090 IF INT(R/2)=R/2 THEN R=R-1
1100 FOR L=1 TO R
1110 READ S$

```

```
1120 NEXT L
1130 PRINT:PRINT
1140 PRINT"WHAT IS THE"
1150 PRINT"ATOMIC NUMBER OF"
1160 PRINT S$
1170 READ C$
1180 INPUT D$
1190 PRINT
1200 IF D$=C$ THEN PRINT"CORRECT"
      :GOTO 1220
1210 PRINT"WRONG"
1220 PRINT C$;"IS THE ATOMIC NUMBER FOR "
      ;S$
1230 RESTORE
1240 PRINT:PRINT
1250 GOTO 1070
```


Programs for the Home

Interest On Money

Now you can analyze the potential of your investments. Whether they be savings accounts or other kinds of interest-bearing investments, the computer will find the money you need now to meet specific future goals. Or money to be realized later from specific investment levels today.

The choices include simple interest and compound interest. You select simple interest or the deposit doubler feature.

In deposit doubler, you can estimate how long it will take your money to double in value at a specific interest rate. You tell the computer the interest rate and it will tell you how long for money to double.

Program Listing

```
10 HOME: CLEAR: PRINT CHR$(7)
20 M$ = "INTEREST ON MONEY"
30 LM = LEN(M$)
40 LT = M + 4
50 FOR L = 1 TO LT
60 AS$ = AS$ + "*"
70 NEXT L
80 P = (40 - LT) / 2
90 HOME
100 PRINT TAB(P) AS$
110 PRINT TAB(P) "*" "; M$; " *"
120 PRINT TAB(P) AS$
130 FOR L = 1 TO 9: PRINT: NEXT L
140 PRINT TAB(9) "PRESS ANY KEY TO START"
145 FOR L = 1 TO 9: PRINT: NEXT L
150 GET KY$
160 HOME: PRINT CHR$(7)
170 PRINT "DO YOU WANT"
180 PRINT "SIMPLE OR COMPOUND INTEREST ?"
190 PRINT: PRINT "PRESS S OR C"
200 A = PEEK(-16384): POKE -16368, 0
210 IF A < 195 OR A > 211 THEN 200
220 IF A = 211 THEN 250
230 IF A = 195 THEN 620
```

```

240 GOTO 200
250 HOME:PRINT CHR$(7)
260 PRINT"FOR":PRINT"MONEY PER YEAR",
    "PRESS Y"
270 PRINT"DEPOSIT DOUBLER","PRESS D"
280 A=PEEK(-16384):POKE -16368,0
    :IF A<128 THEN 280
290 IF A=196 THEN 500
300 IF A=217 THEN 320
310 GOTO 280
320 HOME:PRINT CHR$(7)
330 INPUT"PRESENT VALUE $";PV
340 INPUT"ANNUAL INTEREST PERCENT
    RATE ";AI
350 INPUT"NUMBER OF YEARS ";NY
360 ID=.01*AI:TI=PV*ID*NY:B$=" SIMPLE "
370 HOME:PRINT CHR$(7)
380 PRINT"AT ";AI;"%";B$;" INTEREST:"
390 PRINT:PRINT"SAVINGS AT START"
400 B$=" TOTALS "
410 PRINT B$;"$";PV
420 PRINT"AFTER ";NY;" YEARS, INTEREST"
430 PRINTB$;"$";TI
440 PL=PV+TI
450 PRINT"AFTER ";NY;" YEARS, SAVINGS"
460 PRINT B$;"$";PL
470 GOTO 540
500 HOME:PRINT CHR$(7)
510 INPUT"ANNUAL INTEREST PERCENT RATE "
    ;AI:IF AI=0 THEN 510
520 YR=72/AI
530 PRINT"MONEY DOUBLES IN APPROX. "
    ;YR;" YEARS"
540 FOR L=1 TO 9:PRINT:NEXT L
550 PRINT"FOR MORE","PRESS M"
560 PRINT"TO QUIT","PRESS Q"
570 A=PEEK(-16384):POKE -16368,0
580 IF A<128 THEN 570
590 IF A=205 THEN CLEAR:GOTO 160
600 IF A=209 THEN 1200
610 GOTO 570
620 HOME:PRINT CHR$(7)

```

```

630 PRINT"WHICH DO YOU WANT TO COMPUTE:"
640 PRINT"PRESENT VALUE","PRESS P"
650 PRINT"FUTURE VALUE","PRESS F"
660 A=PEEK(-16384):POKE -16368,0
670 IF A<128 THEN 660
680 IF A=208 THEN 710
690 IF A=198 THEN 800
700 GOTO 660
710 HOME:PRINT CHR$(7)
720 INPUT"DESIRED FUTURE VALUE $";FV
730 INPUT"ANNUAL INTEREST PERCENT RATE "
    ;AI
740 INPUT"NUMBER OF YEARS ";NY
    :IF NY=0 THEN 740
750 ID=.01*AI:PV=FV/(1+ID)^NY
    :TI=FV-PV:B$=" COMPOUND "
760 GOTO 370
800 HOME:PRINT CHR$(7)
810 PRINT"COMPOUNDING CHOICES:"
820 PRINT:PRINT"ANNUAL","PRESS A"
830 PRINT"QUARTERLY","PRESS Q"
840 A=PEEK(-16384):POKE -16368,0
850 IF A<128 THEN 840
860 IF A=193 THEN 900
870 IF A=209 THEN 1000
880 GOTO 840
900 HOME:PRINT CHR$(7)
910 INPUT"PRESENT SAVINGS BALANCE ";PV
920 INPUT"ANNUAL INTEREST PERCENT RATE ";AI
930 INPUT"NUMBER OF YEARS ";NY
940 IF NY=0 THEN 930
950 ID=.01*AI:FV=PV*(1+ID)^NY
    :TI=FV-PV:B$=" COMPOUND "
960 GOTO 370
1000 HOME:PRINT CHR$(7)
1010 INPUT"PRESENT SAVINGS BALANCE ";PV
1020 INPUT"ANNUAL INTEREST PERCENT RATE "
    ;AI
1030 INPUT"NUMBER OF YEARS ";NY
1040 IF NY=0 THEN 1030
1050 QR=4*NY>ID=.0025*AI:FV=PV*(1+ID)^QR
    :TI=FV-PV:B$=" QUARTERLY COMPOUNDED "

```



```
1060 GOTO 370
1200 HOME:PRINT CHR$(7)
1210 FOR L=1 TO 12:PRINT:NEXT L
1220 PRINT TAB(15)"THANK YOU"
```

Sample Run

INTEREST ON MONEY
PRESS ANY KEY TO START

H

DO YOU WANT
SIMPLE OR COMPOUND INTEREST ?

PRESS S OR C
C

WHICH DO YOU WANT TO COMPUTE:
PRESENT VALUE PRESS P
FUTURE VALUE PRESS F
F

COMPOUNDING CHOICES:
ANNUAL PRESS A
QUARTERLY PRESS Q
Q

PRESENT SAVINGS BALANCE:
1000 RETURN
ANNUAL INTEREST RATE:
10 RETURN
NUMBER OF YEARS:
10 RETURN

AT 10% QUARTERLY COMPOUNDED INTEREST:

SAVINGS AT START
TOTALS \$1000

AFTER 10 YEARS, INTEREST
TOTALS \$1685.06387

AFTER 10 YEARS, SAVINGS
TOTALS \$2685.06387

FOR MORE PRESS M
TO QUIT PRESS Q

Club Member List

Now you can keep the names of the members of your club, class, office, or other organization in properly sorted order with this easy to key-in-and-use program.

It causes the Apple computer to store and sort your membership list, and display it alphabetically.

The program accepts up to 100 names and sorts them into alphabetical order. The size of the list is controlled by the DIM statement in line 20.

Note: the screen goes blank while sorting so you must wait. The more names to sort, the longer it takes. If you wish you can change names to something other than names of people, such as items, products, units, etc.

Program Listing

```
10 HOME
20 DIM M$(100)
30 FOR L=1 TO 100
40 INPUT "NAME: ";M$(L)
50 NEXT L
60 HOME
70 T=0
80 FOR L=1 TO 99
90 IF M$(L)<=M$(L+1) THEN 110
100 E$=M$(L):M$(L)=M$(L+1):M$(L+1)=E$
    :T=1
110 NEXT L
120 IF T=1 THEN 70
130 FOR L=1 TO 100
140 IF M$(L)<>" " THEN PRINT M$(L)
150 NEXT L
```

Sample Run

```
RUN
NAME
SMITH ALBERT 1234 MAIN ST NEW YORK NY 10006
NAME
JONES RONALD 99 ORANGE CT SONOMA CA 98765
NAME
BROWN EDWARD 456 44TH ST AKRON OH 45678
NAME
GARDNER EDNA 8 FLOWER AVE ALTOONA PA 16601
NAME
X

BROWN EDWARD 456 44TH ST AKRON OH 45678
GARDNER EDNA 8 FLOWER AVE ALTOONA PA 16601
JONES RONALD 99 ORANGE CT SONOMA CA 98765
SMITH ALBERT 1234 MAIN ST NEW YORK NY 10006
X
```

Electric Bill Analysis

Confused by your electric bill? Think you're being stung? Can't figure when you used all that juice? Now you can analyze your bills for a one-year period and see where you are draining the local utility.

The computer will report annual totals, monthly averages of kilowatt hours used, and the cost of those kilowatts. Most folks receive an electric bill every other month and the computer assumes that's the case. It counts on two months of service per bill. If you receive monthly bills and would like to analyze six months' worth, change only the number 2 in line 110 to a 1. If, on the other hand, you want to analyze 12 monthly bills, also change to 12 the number 6 in lines 20, 30, 280, 310, 340, 370, 400, 430, 460, 490, 520, 550, 580, and 610. Also change the word "bimonthly" in line 750.

You can give the computer information from your electric bills in any monthly order. The program sorts the monthly bills into chronological order within one year. Be

sure to spell out fully and correctly the names of the billing months.

Program Listing

```
10 HOME: CLEAR
20 DIM M$(6)
30 FOR L=1 TO 6
40 INPUT "BILLING MONTH: "; MN$
50 IF MN$="" THEN 240
60 INPUT "KILOWATT HOURS: "; KW$
70 INPUT "NUMBER OF DAYS: "; ND$
80 INPUT "UNIT PRICE: "; UP$
90 M$(L)=MN$+" / "+KW$+" KWH / "+ND$+"
  "DAYS / "+ "$"+UP$
100 N=N+1
110 M=M+2
120 K=VAL(KW$)
130 D=VAL(ND$)
140 P=VAL(UP$)
150 IF N=1 THEN LL=K: LM$=M$(L): HH=K
  : HM$=M$(L)
160 IF K<LL THEN LL=K: LM$=M$(L)
170 IF K>HH THEN HH=K: HM$=M$(L)
180 KK=KK+K
190 DD=DD+D
200 MU=KK/M
210 TC=KK*P
220 MC=TC/M
230 NEXT L
240 HOME
250 PRINT
260 PRINT "MONTH/KWH/DAYS/COST"
270 PRINT
280 FOR L=1 TO 6
290 IF LEFT$(M$(L),3)="JAN" THEN PRINT M$(L)
300 NEXT L
310 FOR L=1 TO 6
320 IF LEFT$(M$(L),3)="FEB" THEN PRINT M$(L)
330 NEXT L
340 FOR L=1 TO 6
```

```

350 IF LEFT$(M$(L),3)="MAR" THEN PRINT M$(L)
360 NEXT L
370 FOR L=1 TO 6
380 IF LEFT$(M$(L),3)="APR" THEN PRINT M$(L)
390 NEXT L
400 FOR L=1 TO 6
410 IF LEFT$(M$(L),3)="MAY" THEN PRINT M$(L)
420 NEXT L
430 FOR L=1 TO 6
440 IF LEFT$(M$(L),3)="JUN" THEN PRINT M$(L)
450 NEXT L
460 FOR L=1 TO 6
470 IF LEFT$(M$(L),3)="JUL" THEN PRINT M$(L)
480 NEXT L
490 FOR L=1 TO 6
500 IF LEFT$(M$(L),3)="AUG" THEN PRINT M$(L)
510 NEXT L
520 FOR L=1 TO 6
530 IF LEFT$(M$(L),3)="SEP" THEN PRINT M$(L)
540 NEXT L
550 FOR L=1 TO 6
560 IF LEFT$(M$(L),3)="OCT" THEN PRINT M$(L)
570 NEXT L
580 FOR L=1 TO 6
590 IF LEFT$(M$(L),3)="NOV" THEN PRINT M$(L)
600 NEXT L
610 FOR L=1 TO 6
620 IF LEFT$(M$(L),3)="DEC" THEN PRINT M$(L)
630 NEXT L
640 PRINT
650 PRINT"FOR BILL ANALYSIS, PRESS A"
660 PRINT"TO QUIT, PRESS Q"
670 GET AA$
680 IF AA$="A" THEN 720
700 IF AA$="Q" THEN 1040
710 GOTO 670
720 HOME
730 PRINT:PRINT
740 PRINT"ELECTRIC BILL ANALYSIS"
750 PRINT N;" BILLS RECEIVED, BIMONTHLY "
760 PRINT

```

```

770 PRINT"TOTALS OVER ";M;" MONTHS"
780 PRINT KK;" KWH USED OVER ";M;" MONTHS"
790 PRINT DD;" DAYS IN ";M;" MONTHS"
800 PRINT" $";TC;" COST OVER";M;" MONTHS"
810 PRINT:PRINT
820 PRINT"FOR MONTHLY AVERAGES, PRESS A"
830 GET AM$
850 IF AM$="A" THEN 870
860 GOTO 830
870 HOME
880 PRINT
890 PRINT"AVERAGES PER MONTH"
900 PRINT
910 PRINT MU;" KWH AVERAGE USE"
920 PRINT " $";MC;" AVERAGE COST"
930 PRINT
940 PRINT "LOWEST-USE BILL":PRINT" ";LM$
950 PRINT "HIGHEST-USE BILL":PRINT" ";HM$
960 PRINT
970 PRINT "TO DO ANOTHER SET, PRESS A"
980 PRINT "TO STOP, PRESS S"
990 GET BB$
1010 IF BB$="A" THEN 10
1020 IF BB$="S" THEN 1040
1030 GOTO 990
1040 HOME
1050 PRINT "                THANK YOU"
1060 GOTO 1060

```

Sample Run

RUN RETURN

BILLING MONTH:
 NOVEMBER RETURN
 KILOWATT HOURS:
 1234 RETURN
 NUMBER OF DAYS:
 60 RETURN
 UNIT PRICE:
 .15 RETURN

BILLING MONTH:
JANUARY RETURN
KILOWATT HOURS:
1255 RETURN
NUMBER OF DAYS:
61 RETURN
UNIT PRICE:
.15 RETURN

BILLING MONTH:
MAY RETURN
KILOWATT HOURS:
1015 RETURN
NUMBER OF DAYS:
58 RETURN
UNIT PRICE:
.15 RETURN

BILLING MONTH:
SEPTEMBER RETURN
KILOWATT HOURS:
997 RETURN
NUMBER OF DAYS:
59 RETURN
UNIT PRICE:
.15 RETURN

BILLING MONTH:
JULY RETURN
KILOWATT HOURS:
1001 RETURN
NUMBER OF DAYS:
60 RETURN
UNIT PRICE:
.15 RETURN

BILLING MONTH:
MARCH RETURN
KILOWATT HOURS:
1188 RETURN
NUMBER OF DAYS:

61 RETURN
UNIT PRICE:
.15 RETURN

MONTH/KWH/DAYS/COST

JANUARY/1255KWH/61DAYS/\$.15
MARCH/1188KWH/61DAYS/\$.15
MAY/1015KWH/58DAYS/\$.15
JULY/1001KWH/60DAYS/\$.15
SEPTEMBER/997KWH/59DAYS/\$.15
NOVEMBER/1234KWH/60DAYS/\$.15

FOR BILL ANALYSIS, PRESS A
TO QUIT, PRESS Q
A

ELECTRIC BILL ANALYSIS
6 BILLS RECEIVED, BIMONTHLY
TOTALS OVER 12 MONTHS
6690 KWH USED OVER 12 MONTHS
359 IN 12 MONTHS
\$1003.50 COST OVER 12 MONTHS
FOR MONTHLY AVERAGES, PRESS A
A

AVERAGES PER MONTH
557.5 KWH AVERAGE USE
\$83.625 AVERAGE COST

LOWEST-USE BILL:
SEPTEMBER/997KWH/59DAYS/\$.15
HIGHEST-USE BILL:
JANUARY/1255KWH/61DAYS/\$.15

TO DO ANOTHER SET, PRESS A
TO STOP, PRESS S
S

THANK YOU

High/Low Bowling Score

Suppose you bowl with a group of friends, each with a different score or set of scores? This program accepts their names and scores and sorts out the persons with the highest and the lowest bowling scores.

Here's how it works: at line 20 the program is dimensioned to hold data on 100 persons. Lines 30 to 130 take in the info on each person. As each person's score is entered, lines 100 to 120 determine if that score is higher or lower than all previously-entered scores. If higher or lower, it is so noted.

To complete data entry, simply press RETURN without data. That prompts your Apple, via lines 140 and 150, to print the lowest score and the highest score.

Naturally, this kind of sorting could be applied to any game with ranges of scores among different plays.

Program Listing

```
10 HOME
20 DIM M$(100)
30 FOR L=1 TO 100
40 INPUT"NAME: ";MN$
50 IF MN$="" THEN 140
60 INPUT"SCORE: ";KW$
70 M$(L)=MN$+" "+KW$
80 N=N+1
90 K=VAL(KW$)
100 IF N=1 THEN LL=K:LM$=M$(L)
    :HH=K:HM$=M$(L)
110 IF K<LL THEN LL=K:LM$=M$(L)
120 IF K>HH THEN HH=K:HM$=M$(L)
130 NEXT L
140 PRINT"LOWEST SCORE: ";LM$
150 PRINT"HIGHEST SCORE: ";HM$
```

Sample Run

```
RUN RETURN
NAME:
JOHN RETURN
```

SCORE: 50

NAME:
SUSAN RETURN
SCORE:
89 RETURN

NAME:
SCOTT RETURN
SCORE:
72 RETURN

NAME:
BOB RETURN
SCORE:
67 RETURN

NAME:
TOM RETURN
SCORE:
87 RETURN

NAME:
RETURN

LOWEST SCORE: JOHN 50
HIGHEST SCORE: SUSAN 89

Funny similes

Give these newfangled gadgets an inch and they'll take a mile. In the case of the computer, give it some tacky retorts and it will spew out an endless string of dumb remarks.

The fun is in having the computer randomly select various words and combine them to make silly sayings.

The random number is used to match the words into similes.

Program Listing

```
10 HOME
20 DATA SHORT,TALL,FAT,LEAN,CLEAN
30 DATA DIRTY,GOOD,BAD,HAPPY,SAD
40 DATA GREEN,RED,YELLOW,BLUE,UGLY
50 DATA PRETTY,SHARP,DULL,TACKY,NATTY
60 DATA STRONG,WEAK,MEAN,NICE,DUMB
70 DATA GNOME,TREE,PIG,BOX,CLOCK
80 DATA TURKEY,GOLD,APPLE,DOG,ROOKIE
90 DATA BEET,BIRD,SKY,SIN,PEACH
100 DATA TACK,RAZOR,PIN,PLUG,BULL
110 DATA WORM,LION,LAMB,PUPPY,OX
200 PRINT CHR$(7)
210 INPUT "WHOM ARE WE DESCRIBING ?";B$
300 T=INT(100*(RND(1)))
310 IF T<1 OR T>25 THEN GOTO 300
320 FOR L=1 TO T
330 READ D$
340 NEXT L
350 RESTORE
400 T=INT(100*(RND(1)))
410 IF T<26 OR T>50 THEN GOTO 400
420 FOR L=1 TO T
430 READ E$
440 NEXT L
450 RESTORE
500 PRINT CHR$(7)
510 PRINT B$;" IS ";D$;" AS A ";E$
520 GOTO 300
```

Recall Tester

How long a number can you remember? As digits are added to lengthen a number, can you remember what you saw on your Apple's video screen?

As the game starts, the computer asks, "How long a number can you recall?" It tells you to "Watch closely."

Then it displays a "short" number. You repeat the short number into the computer. If you are wrong, it gets you to try again. If you were right, it gives you a longer number.

The game goes on until you can't correctly recall a long number in five tries.

If you wish to end the game, enter X rather than a number. The computer will report the length of the number you couldn't get and how many tries you took.

Program Listing

```
10 HOME: CLEAR
20 PRINT CHR$(7)
30 PRINT "HOW LONG A NUMBER CAN YOU
  RECALL ?"
40 PRINT "WATCH CLOSELY..."
50 FOR T=1 TO 1500:NEXT T
100 GOSUB 500
110 K$=K$+N$
120 PRINT:PRINT K$
130 FOR T=1 TO 500:NEXT T
140 HOME:A=A+1
150 IF A=6 THEN 400
160 PRINT "THIS IS YOUR ATTEMPT NUMBER ";A
170 INPUT "WHAT WAS THAT NUMBER ?";G$
180 IF G$="X" THEN 400
190 IF G$=K$ THEN 300
200 PRINT "WRONG. TRY AGAIN."
210 GOTO 120
300 PRINT:PRINT "YES, ";K$;" WAS THE
  NUMBER"
310 PRINT "NOW TRY A LONGER ONE..."
320 FOR T=1 TO 2000:NEXT T
330 A=0:GOTO 100
400 HOME:PRINT CHR$(7)
410 PRINT "YOU GOT OUT AS FAR AS "
  ;D;" DIGITS"
420 PRINT:PRINT "THANKS FOR TRYING"
430 END
500 N=INT(10*(RND(1)))
510 IF N<1 THEN 500
520 N$=STR$(N)
530 D=D+1
540 RETURN
```

Event Timer

Place your computer in a corner and let it time your next chess match. Three-minute egg. Final exam.

The computer asks how many minutes you want for the event you are timing, and then it sounds a bell when the time has passed.

You can calibrate the clock by changing the value of SP in line 10. A larger number will slow down the clock. A smaller value for SP will speed up the clock. As you can see we have started with an SP value of 16. That works just about right on our 48K Apple II Plus, but it is not necessarily right for your machine.

If you want to time an event of less than one minute, use a decimal. For instance, when you want to time a 30-second event, respond to the computer's inquiry with .5 or for 45 seconds key in .75. Use .17 for 10 seconds; .25 for 15 seconds.

Program Listing

```
10 HOME:SP=16
20 PRINT"EVENT TIMER":PRINT
30 PRINT"HOW MANY MINUTES"
40 INPUT"TO THE END OF THE EVENT ";LT
50 PRINT:PRINT"PRESS ANY KEY"
60 PRINT"TO START TIMING"
70 GET ST$
100 HOME
110 C=C+1
120 IF C>(SP*LT*60) THEN 300
130 MN=INT(C/SP/60)
140 SC=INT((C/SP)-(60*MN))
150 PRINT MN;" MINUTES",SC;" SECONDS"
160 GOTO 110
300 HOME:PRINT CHR$(7)
310 PRINT"TIME IS UP"
320 PRINT LT;" MINUTES HAVE PASSED"
330 PRINT:PRINT"TO TIME AGAIN, PRESS
    ANY KEY"
340 CLEAR:GET TA$
350 GOTO 10
```

Photography: Flash Exposure

Use your computer to help take better pictures!

The most important factor in pictures shot with flash is the distance from your flash to the subject. Subjects which are close to you will receive a lot of light while subjects farther away will receive less light.

Check your data sheet for the film you are using. Look for the film guide number. Next, make an estimate of the distance in feet from the flash to your subject.

This program determines the proper *f*/stop setting for your camera. By the way, if the computer tells you to use an *f*/stop setting between two *f*/numbers available on your camera, set your lens opening at the nearest *f*/number or halfway between the two, whichever is closest.

For example, suppose your film has a guide number of 80 and you estimate the flash-to-subject distance at 10 feet. Use *f*/8 on your lens.

Program Listing

```
10 HOME:PRINT CHR$(7)
20 PRINT TAB(13)"FLASH EXPOSURE"
30 PRINT:PRINT
40 INPUT"WHAT IS FILM GUIDE NUMBER ";G
50 INPUT"WHAT IS FLASH-TO-SUBJECT
   DISTANCE ";D
60 F=G/D
70 HOME:PRINT CHR$(7)
80 PRINT"GUIDE NUMBER:",G
90 PRINT"DISTANCE:",D;" FEET"
100 PRINT"SHOOT AT:", "F/";F
110 FOR L=1 TO 9:PRINT:NEXT L
120 PRINT"FOR MORE, PRESS ANY KEY"
130 GET KY$
140 CLEAR:GOTO 30
```

Sample Run

FLASH EXPOSURE

WHAT IS FILM GUIDE NUMBER

```
80          RETURN
WHAT IS FLASH-TO-SUBJECT DISTANCE
10          RETURN
```

```
GUIDE NUMBER: 80
DISTANCE:      10 FEET
SHOOT AT:      F/8
```

FOR MORE, PRESS ANY KEY

Photography: Close Ups

For copying and other close-up work with your camera, you extend the camera lens by using bellows or extension tubes. In doing that, you must allow for an effective increase in the normal *f*/number or your picture will be underexposed.

You make such an exposure compensation whenever the subject distance is less than eight times the focal length of your lens.

This program provides a convenient means of determining the effective *f*/number. For example, if the focal length of your camera is 50mm and the lens-to-film distance (focal length plus extension from infinity position) is 100mm, and the normal *f*/stop would be 22, the corrected stop would be *f*/11.

Or, if you are using a 25mm lens, with 50mm lens-to-film distance, a normal *f*/stop of 8 should be corrected to *f*/4. Be sure to keep both focal length and distance in either mm or inches. Don't mix apples and oranges.

Program Listing

```
10 HOME:PRINT CHR$(7)
20 PRINT TAB(15) "CLOSE UPS"
30 PRINT:PRINT
40 INPUT"WHAT IS NORMAL F/ NUMBER ";F
50 INPUT"WHAT IS LENS-TO-FILM DISTANCE
   IN MM ";D
```

```

60 IF D=0 THEN 50
70 INPUT"WHAT IS THE LENS FOCAL LENGTH
   IN MM ";L
80 N=F*L/D
100 PRINT:PRINT CHR$(7)
110 PRINT"EFFECTIVE F/ NUMBER IS F/";N
200 FOR L=1 TO 9:PRINT:NEXT L
210 PRINT"FOR MORE, PRESS ANY KEY"
220 GET KY$
230 CLEAR:GOTO 30

```

Sample Run

CLOSE UPS

```

WHAT IS NORMAL F/ NUMBER
22          RETURN
WHAT IS LENS-TO-FILM DISTANCE IN MM
100         RETURN
WHAT IS LENS FOCAL LENGTH IN MM
50          RETURN

```

EFFECTIVE F/ NUMBER IS F/11

FOR MORE, PRESS ANY KEY

Savings Quickie

Want a quick idea of how much your savings account will grow over the years? This program is fast to load and speedy to run.

The computer will ask for initial savings balance, annual interest percentage rate, and number of years. In return, it computes compound interest and displays the savings balance at the end of each year in a handy list.

Program Listing

```

10 HOME: CLEAR
20 INPUT"PRESENT SAVINGS BALANCE";B
30 INPUT"INTEREST RATE";I

```



```
40 INPUT"NUMBER OF YEARS";Y
50 FOR L=1 TO Y
60 Z=Z+I*(Z+B)/100
70 PRINT L,Z+B
80 NEXT L
90 END
```

Sample Run

RUN RETURN

PRESENT SAVINGS BALANCE:

652 RETURN

INTEREST RATE?

8 RETURN

NUMBER OF YEARS?

11 RETURN

1 704.16

2 760.4928

3 821.332225

4 887.038802

5 958.001906

6 1034.64206

7 1117.41342

8 1206.8065

9 1303.35102

10 1407.6191

11 1520.22863

Random Number Quality Checker

Ever wonder just how unintentional, haphazard, or unrelated your random numbers are? This program reinforces your confidence in the pseudorandom number generator in the Color Computer.

It causes the machine to generate 100 numbers between zero and 100 and reports how many are above 49 and how many are below 50.

Just for fun, we've thrown in an executive decision maker. That is, the board of directors voted 47 yes, 53 no. Can you imagine it?

Program Listing

```
10 HOME: CLEAR
20 FOR L=1 TO 100
30 X=INT(100*(RND(1)))
40 IF X<50 THEN Y=Y+1
50 IF X>49 THEN N=N+1
60 NEXT L
70 PRINT Y;"YES"
80 PRINT N;"NO"
90 N=0:Y=0
100 PRINT""
110 GOTO 20
```

Sample Run

RUN RETURN

49 YES
51 NO

50 YES
50 NO

49 YES
51 NO

52 YES
48 NO

51 YES
49 NO

47 YES
53 NO

48 YES
52 NO

47 YES
53 NO

Appendix A:

BASIC Words in the APPLE

Here is a handy list of all the words in Applesoft BASIC. The computer uses only one byte of memory space to store each full word of these instructions. That compares with one byte per letter, number or symbol for other information you type into your APPLE. For example, the instruction RETURN is stored all in one byte of memory while a name such as White, for instance, takes up five bytes of memory space.

Generally, the ampersand and the instruction XPLOT are not useful. COLOR, HCOLOR, SCALE, SPEED and ROT must be followed immediately by an equal sign. SCRN, SPC and TAB must be followed immediately by a left parenthesis. HIMEM and LOMEM must be followed immediately by a colon. ATN must have no space between its letters. The instruction TO must not be preceded by the letter A or the computer will think you want the word AT.

&

ABS
AND
ASC
AT
ATN
CALL
CHR\$
CLEAR
COLOR=
CONT
COS
DATA
DEF
DEL
DIM
DRAW
END
EXP

FLASH
FN
FOR
FRE
GET
GOSUB
GOTO
GR
HCOLOR=
HGR
HGR2
HIMEM:
HLIN
HOME
HPLOT
HTAB
IF
IN#
INPUT
INT
INVERSE
LEFT\$
LEN
LET
LIST
LOAD
LOG
LOMEM:
MID\$
NEW
NEXT
NORMAL
NOT
NOTRACE
ON
ONERR
OR
PDL
PEEK
PLOT

POKE
POP
POS
PRINT
PR#

READ
RECALL
REM
RESTORE
RESUME
RETURN
RIGHT\$
RND
ROT=
RUN

SAVE
SCALE=
SCRN(
SGN
SHLOAD
SIN
SPC(
SPEED=
SQR
STEP
STOP
STORE
STR\$

TAB(
TAN
TEXT
THEN
TO
TRACE

USR

VAL
VLIN
VTAB

WAIT

XPLOT
XDRAW

Appendix B: Error Messages

Your APPLE will tell you what went wrong when you make a programming mistake. Here are the messages you might receive:

?CAN'T CONTINUE ERROR

You are incorrectly trying to continue running a program after an error has occurred, or after a line has been deleted, or added to a program. Or the program does not exist.

?DIVISION BY ZERO ERROR

You can't divide a number by zero as you are attempting.

?FORMULA TOO COMPLEX ERROR

You are trying to use more than two IF/THEN instructions at once.

?ILLEGAL DIRECT ERROR

When you try to use your APPLE like a calculator, you are using it in the *immediate execution* mode. When you try to run a previously-entered program, you use the *deferred execution* mode. You cannot use the instructions INPUT, DEF FN, GET or DATA in the *immediate* mode. This error message says you are incorrectly trying to do that.

?ILLEGAL QUANTITY ERROR

You are trying to use a number beyond the range of ability of your APPLE.

?NEXT WITHOUT FOR ERROR

Your NEXT has no matching FOR.

?OUT OF DATA ERROR

You are trying to READ but all DATA has been used.

?OUT OF MEMORY ERROR

Your program is too large. Or you have used too many variables. Or your loops or subroutines nesting is too complex for the APPLE to handle.

?OVERFLOW ERROR

The result of a computation is just too large for the APPLE.

?REDIM'D ARRAY ERROR

You are trying incorrectly to redimension an array you already have dimensioned.

?RETURN WITHOUT GOSUB ERROR

You are trying to use the RETURN instruction but you didn't GOSUB in the first place. RETURN must always be associated with a GOSUB instruction.

?STRING TOO LONG ERROR

You are trying to create a string of more than 255 characters.

?BAD SUBSCRIPT ERROR

You are trying to use an array bigger than the dimension you previously set.

?SYNTAX ERROR

You are not using BASIC correctly. Your punctuation or letters or symbols are wrong.

?TYPE MISMATCH ERROR

You're trying to match apples with oranges. You are trying to mix numeric and string variables. The computer is looking for a numeric value and you are giving it a string or else it wants a string and you are giving it a numeric value.

?UNDEF'D STATEMENT ERROR

You are trying to jump to a nonexistent line. Check your GOTO, GOSUB or THEN instruction.

?UNDEF'D FUNCTION ERROR

You are trying to use a “user defined” function but you haven't previously defined such a function.

books from ARCsoft Publishers

For the TRS-80 PC-1, PC-2, Sharp PC-1211, PC-1500:

99 Tips & Tricks for the New Pocket Computers (PC-2/PC-1500)		
128 pages	\$7.95	ISBN 0-86668-019-5
101 Pocket Computer Programming Tips & Tricks		
128 pages	\$7.95	ISBN 0-86668-004-7
Pocket Computer Programming Made Easy		
128 pages	\$8.95	ISBN 0-86668-009-8
Murder In The Mansion and Other Computer Adventures		
96 pages	\$6.95	ISBN-0-86668-501-4
50 Programs in BASIC for Home, School & Office		
96 pages	\$9.95	ISBN 0-86668-502-2
50 MORE Programs in BASIC for Home, School & Office		
96 pages	\$9.95	ISBN 0-86668-003-9
35 Practical Programs for the Casio Pocket Computer		
96 pages	\$8.95	ISBN 0-86668-014-4
Pocket-BASIC Coding Form programming worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-801-3

For the TRS-80 Color Computer:

101 Color Computer Programming Tips & Tricks		
128 pages	\$7.95	ISBN 0-86668-007-1
55 Color Computer Programs for Home, School & Office		
128 pages	\$9.95	ISBN 0-86668-005-5
55 MORE Color Computer Programs for Home, School & Office		
112 pages	\$9.95	ISBN 0-86668-008-X
Color Computer Graphics		
128 pages	\$9.95	ISBN 0-86668-012-8
The Color Computer Songbook		
96 pages	\$7.95	ISBN 0-86668-011-X
My Buttons Are Blue and Other Love Poems		
96 pages	\$4.95	ISBN 0-86668-013-6
Color Computer BASIC Coding Form program worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-802-1

For the APPLE Computer:

101 APPLE Computer Programming Tips & Tricks		
128 pages	\$8.95	ISBN 0-86668-015-2
33 New APPLE Computer Programs for Home, School & Office		
96 pages	\$8.95	ISBN 0-86668-016-0
APPLE Computer BASIC Coding Form program worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-803-X

For the ATARI 400 and 800 computers:

101 ATARI Computer Programming Tips & Tricks		
128 pages	\$8.95	ISBN 0-86668-022-5
31 New ATARI Computer Programs for Home, School & Office		
96 pages	\$8.95	ISBN 0-86668-018-7
ATARI Computer BASIC Coding Form program worksheet tablets		
40-sheet pad	\$2.95	ISBN 0-86668-806-4

books from ARCsoft Publishers

For the Sinclair ZX-81 and TIMEX/Sinclair 1000 computers:

101 TIMEX 1000 and Sinclair ZX-81 Programming Tips & Tricks		
128 pages	\$7.95	ISBN 0-86668-020-9
37 TIMEX 1000 & Sinclair ZX-81 Programs for Home, School & Office		
96 pages	\$8.95	ISBN 0-86668-021-7
TIMEX 1000 & Sinclair ZX-81 BASIC Coding Form program worksheets		
40-sheet pad	\$2.95	ISBN 0-86668-807-2

For the electronics hobbyist:

25 Quick-N-Easy Electronics Projects		
96 pages	\$4.95	ISBN 0-86668-023-3
25 Electronics Projects for Beginners		
96 pages	\$4.95	ISBN 0-86668-017-9
25 Easy-To-Build One-Night & Weekend Electronics Projects		
96 pages	\$4.95	ISBN 0-86668-010-1

Computer program-writing worksheets:

Each in tablet of 40 sheets with stiff backing

Universal BASIC Coding Form	\$2.95
IBM Personal Computer BASIC Coding Form	\$2.95
ATARI Computer BASIC Coding Form	\$2.95
Pocket Computer BASIC Coding Form	\$2.95
APPLE Computer BASIC Coding Form	\$2.95
Color Computer BASIC Coding Form	\$2.95
TIMEX 1000/Sinclair ZX-81 BASIC Form	\$2.95

ISBN: International Standard Book Number

ARCsoft Publishers

**Post Office Box 132
Woodsboro, Maryland 21798**

33 New APPLE Computer Programs for Home, School & Office

by Fred White

For the small-business office, classroom, home, a brand-new volume of tested ready-to-run software in BASIC for the APPLE computer.

For businessmen, *Invoice Computer*, *Inventory Review*, *Pricing Mark-Up*, *Profit Estimator*, daily codes, hourly wages and much more.

For students and teachers, *Math Flasher*, exam score sorting, U.S. presidents, foreign capitals, weights & measures, lakes of the world, chemistry, and many more.

For the home, electric bill analysis, event timer, photography, interest on money, club member list, and more.

This is a complete set of software. All you have to do is type it in and your APPLE computer will be ready to run. Use programs alone or in combination to make your computer do what you want. Using these programs, you'll learn what to do when programming and operating your APPLE; how to do it; when to make changes and when not to. All from this exclusive source. As with all books from *ARCsoft Publishers*, these programs are simple to enter, easy to use and provide hours of operating practicality, usefulness and fun.

There are 35 fresh BASIC programs in this book plus an Appendix of all APPLE words in BASIC and APPLE error messages. With minor modifications to program lines, and changes to some graphics commands, each program will run on TRS-80 Model I, II, III, 16, Color, Pocket, Pet, Vic, Atari, Sinclair, Timex, Sharp, Casio, Panasonic, Quasar and any other micro-computer using the BASIC programming language. Even if you don't own a computer yet, learn programming by reading these listings and following their logic.

ARCsoft Publishers

WOODSBORO, MARYLAND 21798

